
CHAPTER

ONE

HIGH-LEVEL INTERFACE

Valuations can be defined conveniently on some Sage rings such as p-adic rings and function fields.

1.1 p-adic valuations

Valuations on number fields can be easily specified if they uniquely extend the valuation of a rational prime:

sage: v = QQ.valuation(2)
sage: v(1024)
10

They are normalized such that the rational prime has valuation 1:

sage: K.<a> = NumberField(x^2 + x + 1)
sage: v = K.valuation(2)
sage: v(1024)
10

If there are multiple valuations over a prime, they can be obtained by extending a valuation from a smaller ring:

sage: K.<a> = NumberField(x^2 + x + 1)
sage: K.valuation(7)
Traceback (most recent call last):
...
ValueError: The valuation Gauss valuation induced by 7-adic valuation does not␣
→˓approximate a unique extension of 7-adic valuation with respect to x^2 + x + 1
sage: w,ww = QQ.valuation(7).extensions(K)
sage: w(a + 3), ww(a + 3)
(1, 0)
sage: w(a + 5), ww(a + 5)
(0, 1)

1.2 Valuations on Function Fields

Similarly, valuations can be defined on function fields:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(x)
sage: v(1/x)
-1

sage: v = K.valuation(1/x)
sage: v(1/x)
1

On extensions of function fields, valuations can be created by providing a prime on the underlying rational function field
when the extension is unique:

sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = L.valuation(x)
sage: v(x)
1

Valuations can also be extended from smaller function fields:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(x - 4)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v.extensions(L)
[[ (x - 4)-adic valuation, v(y + 2) = 1 ]-adic valuation,
[ (x - 4)-adic valuation, v(y - 2) = 1 ]-adic valuation]

2 Chapter 1. High-Level Interface



CHAPTER

TWO

LOW-LEVEL INTERFACE

2.1 Mac Lane valuations

Internally, all the above is backed by the algorithms described in [Mac1936I] and [Mac1936II]. Let us consider the
extensions of K.valuation(x - 4) to the field 𝐿 above to outline how this works internally.

First, the valuation on𝐾 is induced by a valuation on2[𝑥]. To construct this valuation, we start from the trivial valuation
on
𝑄 and consider its induced Gauss valuation on
𝑄[𝑥], i.e., the valuation that assigns to a polynomial the minimum of the coefficient valuations:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ))

The Gauss valuation can be augmented by specifying that 𝑥− 4 has valuation 1:

sage: v = v.augmentation(x - 4, 1); v
[ Gauss valuation induced by Trivial valuation on Rational Field, v(x - 4) = 1 ]

This valuation then extends uniquely to the fraction field:

sage: K.<x> = FunctionField(QQ)
sage: v = v.extension(K); v
(x - 4)-adic valuation

Over the function field we repeat the above process, i.e., we define the Gauss valuation induced by it and augment it to
approximate an extension to 𝐿:

sage: R.<y> = K[]
sage: w = GaussValuation(R, v)
sage: w = w.augmentation(y - 2, 1); w
[ Gauss valuation induced by (x - 4)-adic valuation, v(y - 2) = 1 ]
sage: L.<y> = K.extension(y^2 - x)
sage: ww = w.extension(L); ww
[ (x - 4)-adic valuation, v(y - 2) = 1 ]-adic valuation

3

2.2 Limit valuations

In the previous example the final valuation ww is not merely given by evaluating w on the ring𝐾[𝑦]:

sage: ww(y^2 - x)
+Infinity
sage: y = R.gen()
sage: w(y^2 - x)
1

Instead ww is given by a limit, i.e., an infinite sequence of augmentations of valuations:

sage: ww._base_valuation
[ Gauss valuation induced by (x - 4)-adic valuation, v(y - 2) = 1 , … ]

The terms of this infinite sequence are computed on demand:

sage: ww._base_valuation._approximation
[ Gauss valuation induced by (x - 4)-adic valuation, v(y - 2) = 1 ]
sage: ww(y - 1/4*x - 1)
2
sage: ww._base_valuation._approximation
[ Gauss valuation induced by (x - 4)-adic valuation, v(y + 1/64*x^2 - 3/8*x - 3/4) =␣
→˓3 ]

2.3 Non-classical valuations

Using the low-level interface we are not limited to classical valuations on function fields that correspond to points on the
corresponding projective curves. Instead we can start with a non-trivial valuation on the field of constants:

sage: v = QQ.valuation(2)
sage: R.<x> = QQ[]
sage: w = GaussValuation(R, v) # v is not trivial
sage: K.<x> = FunctionField(QQ)
sage: w = w.extension(K)
sage: w.residue_field()
Rational function field in x over Finite Field of size 2

4 Chapter 2. Low-Level Interface



CHAPTER

THREE

MAC LANE APPROXIMANTS

The main tool underlying this package is an algorithm by Mac Lane to compute, starting from a Gauss valuation on a
polynomial ring and a monic squarefree polynomial G, approximations to the limit valuation which send G to infinity:

sage: v = QQ.valuation(2)
sage: R.<x> = QQ[]
sage: f = x^5 + 3*x^4 + 5*x^3 + 8*x^2 + 6*x + 12
sage: v.mac_lane_approximants(f) # random output (order may vary)
[[ Gauss valuation induced by 2-adic valuation, v(x^2 + x + 1) = 3 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 1/2 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 1 ]]

From these approximants one can already see the residual degrees and ramification indices of the corresponding exten-
sions. The approximants can be pushed to arbitrary precision, corresponding to a factorization of f:

sage: v.mac_lane_approximants(f, required_precision=10) # random output
[[ Gauss valuation induced by 2-adic valuation, v(x^2 + 193*x + 13/21) = 10 ],
[ Gauss valuation induced by 2-adic valuation, v(x + 86) = 10 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 1/2, v(x^2 + 36/11*x + 2/17) =␣
→˓11 ]]

5

CHAPTER

FOUR

REFERENCES

The theory was originally described in [Mac1936I] and [Mac1936II]. A summary and some algorithmic details can also
be found in Chapter 4 of [Rüt2014].

CHAPTER

FIVE

MORE DETAILS

5.1 Value groups of discrete valuations

This file defines additive sub(semi-)groups of 2 and related structures.

AUTHORS:

• Julian Rüth (2013-09-06): initial version

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.value_group()
Additive Abelian Group generated by 1
sage: v.value_semigroup()
Additive Abelian Semigroup generated by 1

class sage.rings.valuation.value_group.DiscreteValuationCodomain

Bases: UniqueRepresentation, Parent

The codomain of discrete valuations, the rational numbers extended by ±∞.

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValuationCodomain
sage: C = DiscreteValuationCodomain(); C
Codomain of Discrete Valuations

class sage.rings.valuation.value_group.DiscreteValueGroup(generator)
Bases: UniqueRepresentation, Parent

The value group of a discrete valuation, an additive subgroup of 2 generated by generator.

INPUT:

• generator – a rational number

Note

Wedo not rely on the functionality provided by additive abelian groups in Sage since these require the underlying
set to be the integers. Therefore, we roll our own Z-module here. We could have used AdditiveAbelian-
GroupWrapper here, but it seems to be somewhat outdated. In particular, generic group functionality should
now come from the category and not from the super-class. A facade of Q appeared to be the better approach.

EXAMPLES:

9

sage: from sage.rings.valuation.value_group import DiscreteValueGroup
sage: D1 = DiscreteValueGroup(0); D1
Trivial Additive Abelian Group
sage: D2 = DiscreteValueGroup(4/3); D2
Additive Abelian Group generated by 4/3
sage: D3 = DiscreteValueGroup(-1/3); D3
Additive Abelian Group generated by 1/3

denominator()

Return the denominator of a generator of this group.

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValueGroup
sage: DiscreteValueGroup(3/8).denominator()
8

gen()

Return a generator of this group.

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValueGroup
sage: DiscreteValueGroup(-3/8).gen()
3/8

index(other)
Return the index of other in this group.

INPUT:

• other – a subgroup of this group

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValueGroup
sage: DiscreteValueGroup(3/8).index(DiscreteValueGroup(3))
8
sage: DiscreteValueGroup(3).index(DiscreteValueGroup(3/8))
Traceback (most recent call last):
...
ValueError: other must be a subgroup of this group
sage: DiscreteValueGroup(3).index(DiscreteValueGroup(0))
Traceback (most recent call last):
...
ValueError: other must have finite index in this group
sage: DiscreteValueGroup(0).index(DiscreteValueGroup(0))
1
sage: DiscreteValueGroup(0).index(DiscreteValueGroup(3))
Traceback (most recent call last):
...
ValueError: other must be a subgroup of this group

is_trivial()

Return whether this is the trivial additive abelian group.

EXAMPLES:

10 Chapter 5. More Details

sage: from sage.rings.valuation.value_group import DiscreteValueGroup
sage: DiscreteValueGroup(-3/8).is_trivial()
False
sage: DiscreteValueGroup(0).is_trivial()
True

numerator()

Return the numerator of a generator of this group.

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValueGroup
sage: DiscreteValueGroup(3/8).numerator()
3

some_elements()

Return some typical elements in this group.

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValueGroup
sage: DiscreteValueGroup(-3/8).some_elements()
[3/8, -3/8, 0, 42, 3/2, -3/2, 9/8, -9/8]

class sage.rings.valuation.value_group.DiscreteValueSemigroup(generators)
Bases: UniqueRepresentation, Parent

The value semigroup of a discrete valuation, an additive subsemigroup of 2 generated by generators.

INPUT:

• generators – rational numbers

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValueSemigroup
sage: D1 = DiscreteValueSemigroup(0); D1
Trivial Additive Abelian Semigroup
sage: D2 = DiscreteValueSemigroup(4/3); D2
Additive Abelian Semigroup generated by 4/3
sage: D3 = DiscreteValueSemigroup([-1/3, 1/2]); D3
Additive Abelian Semigroup generated by -1/3, 1/2

gens()

Return the generators of this semigroup.

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValueSemigroup
sage: DiscreteValueSemigroup(-3/8).gens()
(-3/8,)

is_group()

Return whether this semigroup is a group.

EXAMPLES:

5.1. Value groups of discrete valuations 11

sage: from sage.rings.valuation.value_group import DiscreteValueSemigroup
sage: DiscreteValueSemigroup(1).is_group()
False
sage: D = DiscreteValueSemigroup([-1, 1])
sage: D.is_group()
True

Invoking this method also changes the category of this semigroup if it is a group:

sage: D in AdditiveMagmas().AdditiveAssociative().AdditiveUnital().
→˓AdditiveInverse()
True

is_trivial()

Return whether this is the trivial additive abelian semigroup.

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValueSemigroup
sage: DiscreteValueSemigroup(-3/8).is_trivial()
False
sage: DiscreteValueSemigroup([]).is_trivial()
True

some_elements()

Return some typical elements in this semigroup.

EXAMPLES:

sage: from sage.rings.valuation.value_group import DiscreteValueSemigroup
sage: list(DiscreteValueSemigroup([-3/8,1/2]).some_elements()) #␣
→˓needs sage.numerical.mip
[0, -3/8, 1/2, ...]

5.2 Discrete valuations

This file defines abstract base classes for discrete (pseudo-)valuations.

AUTHORS:

• Julian Rüth (2013-03-16): initial version

EXAMPLES:

Discrete valuations can be created on a variety of rings:

sage: ZZ.valuation(2)
2-adic valuation
sage: GaussianIntegers().valuation(3) #␣
→˓needs sage.rings.number_field
3-adic valuation
sage: QQ.valuation(5)
5-adic valuation
sage: Zp(7).valuation()
7-adic valuation

12 Chapter 5. More Details

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: K.valuation(x)
(x)-adic valuation
sage: K.valuation(x^2 + 1)
(x^2 + 1)-adic valuation
sage: K.valuation(1/x)
Valuation at the infinite place

sage: R.<x> = QQ[]
sage: v = QQ.valuation(2)
sage: w = GaussValuation(R, v)
sage: w.augmentation(x, 3)
[ Gauss valuation induced by 2-adic valuation, v(x) = 3 ]

We can also define discrete pseudo-valuations, i.e., discrete valuations that send more than just zero to infinity:

sage: w.augmentation(x, infinity)
[ Gauss valuation induced by 2-adic valuation, v(x) = +Infinity ]

class sage.rings.valuation.valuation.DiscretePseudoValuation(parent)
Bases: Morphism

Abstract base class for discrete pseudo-valuations, i.e., discrete valuations which might send more that just zero to
infinity.

INPUT:

• domain – an integral domain

EXAMPLES:

sage: v = ZZ.valuation(2); v # indirect doctest
2-adic valuation

is_equivalent(f , g)
Return whether f and g are equivalent.

EXAMPLES:

sage: v = QQ.valuation(2)
sage: v.is_equivalent(2, 1)
False
sage: v.is_equivalent(2, -2)
True
sage: v.is_equivalent(2, 0)
False
sage: v.is_equivalent(0, 0)
True

class sage.rings.valuation.valuation.DiscreteValuation(parent)
Bases: DiscretePseudoValuation

Abstract base class for discrete valuations.

EXAMPLES:

5.2. Discrete valuations 13

sage: v = QQ.valuation(2)
sage: R.<x> = QQ[]
sage: v = GaussValuation(R, v)
sage: w = v.augmentation(x, 1337); w # indirect doctest
[ Gauss valuation induced by 2-adic valuation, v(x) = 1337 ]

is_discrete_valuation()

Return whether this valuation is a discrete valuation.

EXAMPLES:

sage: v = valuations.TrivialValuation(ZZ)
sage: v.is_discrete_valuation()
True

mac_lane_approximant(G, valuation, approximants=None)
Return the approximant from mac_lane_approximants() for G which is approximated by or approx-
imates valuation.

INPUT:

• G – a monic squarefree integral polynomial in a univariate polynomial ring over the domain of this
valuation

• valuation – a valuation on the parent of G

• approximants – the output of mac_lane_approximants(); if not given, it is computed

EXAMPLES:

sage: v = QQ.valuation(2)
sage: R.<x> = QQ[]
sage: G = x^2 + 1

We can select an approximant by approximating it:

sage: w = GaussValuation(R, v).augmentation(x + 1, 1/2)
sage: v.mac_lane_approximant(G, w) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1/2 ]

As long as this is the only matching approximant, the approximation can be very coarse:

sage: w = GaussValuation(R, v)
sage: v.mac_lane_approximant(G, w) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1/2 ]

Or it can be very specific:

sage: w = GaussValuation(R, v).augmentation(x + 1, 1/2).augmentation(G,␣
→˓infinity)
sage: v.mac_lane_approximant(G, w) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1/2 ]

But it must be an approximation of an approximant:

14 Chapter 5. More Details

sage: w = GaussValuation(R, v).augmentation(x, 1/2)
sage: v.mac_lane_approximant(G, w)
Traceback (most recent call last):
...
ValueError: The valuation
[ Gauss valuation induced by 2-adic valuation, v(x) = 1/2 ] is
not an approximant for a valuation which extends 2-adic valuation
with respect to x^2 + 1 since the valuation of x^2 + 1
does not increase in every step

The valuation must single out one approximant:

sage: G = x^2 - 1
sage: w = GaussValuation(R, v)
sage: v.mac_lane_approximant(G, w) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
Traceback (most recent call last):
...
ValueError: The valuation Gauss valuation induced by 2-adic valuation
does not approximate a unique extension of 2-adic valuation
with respect to x^2 - 1

sage: w = GaussValuation(R, v).augmentation(x + 1, 1)
sage: v.mac_lane_approximant(G, w) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
Traceback (most recent call last):
...
ValueError: The valuation
[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1 ] does not
approximate a unique extension of 2-adic valuation with respect to x^2 - 1

sage: w = GaussValuation(R, v).augmentation(x + 1, 2)
sage: v.mac_lane_approximant(G, w) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
[ Gauss valuation induced by 2-adic valuation, v(x + 1) = +Infinity ]

sage: w = GaussValuation(R, v).augmentation(x + 3, 2)
sage: v.mac_lane_approximant(G, w) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1 ]

mac_lane_approximants(G, assume_squarefree=False, require_final_EF=True, required_precision=-1,
require_incomparability=False, require_maximal_degree=False,
algorithm='serial')

Return approximants on𝐾[𝑥] for the extensions of this valuation to 𝐿 = 𝐾[𝑥]/(𝐺).

If 𝐺 is an irreducible polynomial, then this corresponds to extensions of this valuation to the completion of
𝐿.

INPUT:

• G – a monic squarefree integral polynomial in a univariate polynomial ring over the domain of this
valuation

• assume_squarefree – boolean (default: False); whether to assume that G is squarefree. If True,
the squafreeness of G is not verified though it is necessary when require_final_EF is set for the
algorithm to terminate.

• require_final_EF – boolean (default: True); whether to require the returned key polynomials to

5.2. Discrete valuations 15

be in one-to-one correspondance to the extensions of this valuation to L and require them to have the
ramification index and residue degree of the valuations they correspond to.

• required_precision – a number or infinity (default: -1); whether to require the last key polyno-
mial of the returned valuations to have at least that valuation.

• require_incomparability – boolean (default: False); whether to require the returned valu-
ations to be incomparable (with respect to the partial order on valuations defined by comparing them
pointwise.)

• require_maximal_degree – boolean (default: False); whether to require the last key polyno-
mial of the returned valuation to have maximal degree. This is most relevant when using this algorithm
to compute approximate factorizations of G, when set to True, the last key polynomial has the same
degree as the corresponding factor.

• algorithm – one of �serial� or �parallel� (default: �serial�); whether or not to paral-
lelize the algorithm

EXAMPLES:

sage: v = QQ.valuation(2)
sage: R.<x> = QQ[]
sage: v.mac_lane_approximants(x^2 + 1) #␣
→˓needs sage.geometry.polyhedron
[[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1/2 ]]
sage: v.mac_lane_approximants(x^2 + 1, required_precision=infinity) #␣
→˓needs sage.geometry.polyhedron
[[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1/2,

v(x^2 + 1) = +Infinity ]]
sage: v.mac_lane_approximants(x^2 + x + 1)
[[ Gauss valuation induced by 2-adic valuation, v(x^2 + x + 1) = +Infinity ]]

Note that G does not need to be irreducible. Here, we detect a factor 𝑥+ 1 and an approximate factor 𝑥+ 1
(which is an approximation to 𝑥− 1):

sage: v.mac_lane_approximants(x^2 - 1) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
[[ Gauss valuation induced by 2-adic valuation, v(x + 1) = +Infinity ],
[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1 ]]

However, it needs to be squarefree:

sage: v.mac_lane_approximants(x^2)
Traceback (most recent call last):
...
ValueError: G must be squarefree

montes_factorization(G, assume_squarefree=False, required_precision=None)
Factor G over the completion of the domain of this valuation.

INPUT:

• G – a monic polynomial over the domain of this valuation

• assume_squarefree – boolean (default: False); whether to assume G to be squarefree

• required_precision – a number or infinity (default: infinity); if infinity, the returned poly-
nomials are actual factors of G, otherwise they are only factors with precision at least required_pre-
cision.

16 Chapter 5. More Details

ALGORITHM:

We compute mac_lane_approximants() with required_precision. The key polynomials ap-
proximate factors of G. This can be very slow unless required_precision is set to zero. Single factor
lifting could improve this significantly.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: k = Qp(5,4)
sage: v = k.valuation()
sage: R.<x> = k[]
sage: G = x^2 + 1
sage: v.montes_factorization(G) #␣
→˓needs sage.geometry.polyhedron
((1 + O(5^4))*x + 2 + 5 + 2*5^2 + 5^3 + O(5^4))
* ((1 + O(5^4))*x + 3 + 3*5 + 2*5^2 + 3*5^3 + O(5^4))

The computation might not terminate over incomplete fields (in particular because the factors can not be
represented there):

sage: R.<x> = QQ[]
sage: v = QQ.valuation(2)
sage: v.montes_factorization(x^6 - 1) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
(x - 1) * (x + 1) * (x^2 - x + 1) * (x^2 + x + 1)

sage: v.montes_factorization(x^7 - 1) # not tested #␣
→˓needs sage.rings.padics

sage: v.montes_factorization(x^7 - 1, required_precision=5) #␣
→˓needs sage.geometry.polyhedron sage.rings.padics
(x - 1) * (x^3 - 5*x^2 - 6*x - 1) * (x^3 + 6*x^2 + 5*x - 1)

REFERENCES:

The underlying algorithm is described in [Mac1936II] and thoroughly analyzed in [GMN2008].

class sage.rings.valuation.valuation.InfiniteDiscretePseudoValuation(parent)
Bases: DiscretePseudoValuation

Abstract base class for infinite discrete pseudo-valuations, i.e., discrete pseudo-valuations which are not discrete
valuations.

EXAMPLES:

sage: v = QQ.valuation(2)
sage: R.<x> = QQ[]
sage: v = GaussValuation(R, v)
sage: w = v.augmentation(x, infinity); w # indirect doctest
[ Gauss valuation induced by 2-adic valuation, v(x) = +Infinity ]

is_discrete_valuation()

Return whether this valuation is a discrete valuation.

EXAMPLES:

sage: v = QQ.valuation(2)
sage: R.<x> = QQ[]

(continues on next page)

5.2. Discrete valuations 17



Discrete Valuations and Discrete Pseudo-Valuations, Release 10.5.rc0

(continued from previous page)

sage: v = GaussValuation(R, v)
sage: v.is_discrete_valuation()
True
sage: w = v.augmentation(x, infinity)
sage: w.is_discrete_valuation()
False

class sage.rings.valuation.valuation.MacLaneApproximantNode(valuation, parent, ef ,
principal_part_bound,
coefficients, valuations)

Bases: object

A node in the tree computed by DiscreteValuation.mac_lane_approximants().

Leaves in the computation of the tree of approximants mac_lane_approximants(). Each vertex consists
of a tuple (v,ef,p,coeffs,vals) where v is an approximant, i.e., a valuation, ef is a boolean, p is the
parent of this vertex, and coeffs and vals are cached values. (Only v and ef are relevant, everything else
are caches/debug info.) The boolean ef denotes whether v already has the final ramification index E and residue
degree F of this approximant. An edge V – P represents the relation P.v ᅔ V.v (pointwise on the polynomial ring
K[x]) between the valuations.

class sage.rings.valuation.valuation.NegativeInfiniteDiscretePseudoValuation(par-
ent)

Bases: InfiniteDiscretePseudoValuation

Abstract base class for pseudo-valuations which attain the value∞ and−∞, i.e., whose domain contains an element
of valuation∞ and its inverse.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ)).augmentation(x,␣
→˓infinity)
sage: K.<x> = FunctionField(QQ)
sage: w = K.valuation(v)

is_negative_pseudo_valuation()

Return whether this valuation attains the value −∞.

EXAMPLES:

sage: R.<x> = QQ[]
sage: u = GaussValuation(R, valuations.TrivialValuation(QQ))
sage: v = u.augmentation(x, infinity)
sage: v.is_negative_pseudo_valuation()
False
sage: K.<x> = FunctionField(QQ)
sage: w = K.valuation(v)
sage: w.is_negative_pseudo_valuation()
True

18 Chapter 5. More Details



Discrete Valuations and Discrete Pseudo-Valuations, Release 10.5.rc0

5.3 Spaces of valuations

This module provides spaces of exponential pseudo-valuations on integral domains. It currently only provides support for
such valuations if they are discrete, i.e., their image is a discrete additive subgroup of the rational numbers extended by
∞.

AUTHORS:

• Julian Rüth (2016-10-14): initial version

EXAMPLES:

sage: QQ.valuation(2).parent()
Discrete pseudo-valuations on Rational Field

Note

Note that many tests not only in this module do not create instances of valuations directly since this gives the wrong
inheritance structure on the resulting objects:

sage: from sage.rings.valuation.valuation_space import DiscretePseudoValuationSpace
sage: from sage.rings.valuation.trivial_valuation import␣
→˓TrivialDiscretePseudoValuation
sage: H = DiscretePseudoValuationSpace(QQ)
sage: v = TrivialDiscretePseudoValuation(H)
sage: v._test_category()
Traceback (most recent call last):
...
AssertionError: False is not true

Instead, the valuations need to be created through the __make_element_class__ of the containing space:

sage: from sage.rings.valuation.trivial_valuation import␣
→˓TrivialDiscretePseudoValuation
sage: v = H.__make_element_class__(TrivialDiscretePseudoValuation)(H)
sage: v._test_category()

The factories such as TrivialPseudoValuation provide the right inheritance structure:

sage: v = valuations.TrivialPseudoValuation(QQ)
sage: v._test_category()

class sage.rings.valuation.valuation_space.DiscretePseudoValuationSpace(domain)

Bases: UniqueRepresentation, Homset

The space of discrete pseudo-valuations on domain.

EXAMPLES:

sage: from sage.rings.valuation.valuation_space import␣
→˓DiscretePseudoValuationSpace
sage: H = DiscretePseudoValuationSpace(QQ)
sage: QQ.valuation(2) in H
True

5.3. Spaces of valuations 19

../../../../../../../html/en/reference/structure/sage/structure/unique_representation.html#sage.structure.unique_representation.UniqueRepresentation
../../../../../../../html/en/reference/categories/sage/categories/homset.html#sage.categories.homset.Homset


Discrete Valuations and Discrete Pseudo-Valuations, Release 10.5.rc0

Note

We do not distinguish between the space of discrete valuations and the space of discrete pseudo-valuations.
This is entirely for practical reasons: We would like to model the fact that every discrete valuation is also a
discrete pseudo-valuation. At first, it seems to be sufficient to make sure that the in operator works which can
essentially be achieved by overriding_element_constructor_ of the space of discrete pseudo-valuations
to accept discrete valuations by just returning them. Currently, however, if one does not change the parent of
an element in _element_constructor_ to self, then one cannot register that conversion as a coercion.
Consequently, the operators <= and >= cannot be made to work between discrete valuations and discrete
pseudo-valuations on the same domain (because the implementation only calls _richcmp if both operands
have the same parent.) Of course, we could override __ge__ and __le__ but then we would likely run into
other surprises. So in the end, we went for a single homspace for all discrete valuations (pseudo or not) as this
makes the implementation much easier.

Todo

The comparison problem might be fixed by Issue #22029 or similar.

class ElementMethods

Bases: object

Provides methods for discrete pseudo-valuations that are added automatically to valuations in this space.

EXAMPLES:

Here is an example of a method that is automagically added to a discrete valuation:

sage: from sage.rings.valuation.valuation_space import␣
→˓DiscretePseudoValuationSpace
sage: H = DiscretePseudoValuationSpace(QQ)
sage: QQ.valuation(2).is_discrete_pseudo_valuation() # indirect doctest
True

The methods will be provided even if the concrete type is not created with __make_element_class__:

sage: from sage.rings.valuation.valuation import DiscretePseudoValuation
sage: m = DiscretePseudoValuation(H)
sage: m.parent() is H
True
sage: m.is_discrete_pseudo_valuation()
True

However, the category framework advises you to use inheritance:

sage: m._test_category()
Traceback (most recent call last):
...
AssertionError: False is not true

Using __make_element_class__, makes your concrete valuation inherit from this class:

sage: m = H.__make_element_class__(DiscretePseudoValuation)(H)
sage: m._test_category()

change_domain(ring)
Return this valuation over ring.

Unlike extension() or restriction(), this might not be completely sane mathematically. It is
essentially a conversion of this valuation into another space of valuations.

EXAMPLES:

sage: v = QQ.valuation(3)
sage: v.change_domain(ZZ)
3-adic valuation

element_with_valuation(s)
Return an element in the domain of this valuation with valuation s.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.element_with_valuation(10)
1024

extension(ring)
Return the unique extension of this valuation to ring.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: w = v.extension(QQ)
sage: w.domain()
Rational Field

extensions(ring)
Return the extensions of this valuation to ring.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.extensions(QQ)
[2-adic valuation]

inverse(x, precision)
Return an approximate inverse of x.

The element returned is such that the product differs from 1 by an element of valuation at least preci-
sion.

INPUT:
• x – an element in the domain of this valuation
• precision – a rational or infinity

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: x = 3
sage: y = v.inverse(3, 2); y
3
sage: x*y - 1
8

This might not be possible for elements of positive valuation:

sage: v.inverse(2, 2)
Traceback (most recent call last):
...
ValueError: element has no approximate inverse in this ring

Of course this always works over fields:

sage: v = QQ.valuation(2)
sage: v.inverse(2, 2)
1/2

is_discrete_pseudo_valuation()

Return whether this valuation is a discrete pseudo-valuation.

EXAMPLES:

sage: QQ.valuation(2).is_discrete_pseudo_valuation()
True

is_discrete_valuation()

Return whether this valuation is a discrete valuation, i.e., whether it is a discrete pseudo val-
uation that only sends zero to∞.

EXAMPLES:

sage: QQ.valuation(2).is_discrete_valuation()
True

is_negative_pseudo_valuation()

Return whether this valuation is a discrete pseudo-valuation that does attain −∞, i.e., it is non-trivial
and its domain contains an element with valuation∞ that has an inverse.

EXAMPLES:

sage: QQ.valuation(2).is_negative_pseudo_valuation()
False

is_trivial()

Return whether this valuation is trivial, i.e., whether it is constant∞ or constant zero for everything but
the zero element.

Subclasses need to override this method if they do not implement uniformizer().

EXAMPLES:

sage: QQ.valuation(7).is_trivial()
False

lift(X)

Return a lift of X in the domain which reduces down to X again via reduce().

EXAMPLES:

sage: v = QQ.valuation(2)
sage: v.lift(v.residue_ring().one())
1

lower_bound(x)
Return a lower bound of this valuation at x.

Use this method to get an approximation of the valuation of x when speed is more important than
accuracy.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.lower_bound(2^10)
10

reduce(x)
Return the image of x in the residue_ring() of this valuation.

EXAMPLES:

sage: v = QQ.valuation(2)
sage: v.reduce(2)
0
sage: v.reduce(1)
1
sage: v.reduce(1/3)
1
sage: v.reduce(1/2)
Traceback (most recent call last):
...
ValueError: reduction is only defined for elements of nonnegative␣
→˓valuation

residue_field()

Return the residue field of this valuation, i.e., the field of fractions of the residue_ring(), the
elements of nonnegative valuation modulo the elements of positive valuation.

EXAMPLES:

sage: QQ.valuation(2).residue_field()
Finite Field of size 2
sage: valuations.TrivialValuation(QQ).residue_field()
Rational Field

sage: valuations.TrivialValuation(ZZ).residue_field()
Rational Field
sage: GaussValuation(ZZ[�x�], ZZ.valuation(2)).residue_field()
Rational function field in x over Finite Field of size 2

residue_ring()

Return the residue ring of this valuation, i.e., the elements of nonnegative valuation modulo the elements
of positive valuation. EXAMPLES:

sage: QQ.valuation(2).residue_ring()
Finite Field of size 2
sage: valuations.TrivialValuation(QQ).residue_ring()
Rational Field

Note that a residue ring always exists, even when a residue field may not:

sage: valuations.TrivialPseudoValuation(QQ).residue_ring()
Quotient of Rational Field by the ideal (1)
sage: valuations.TrivialValuation(ZZ).residue_ring()
Integer Ring
sage: GaussValuation(ZZ[�x�], ZZ.valuation(2)).residue_ring()
Univariate Polynomial Ring in x over Finite Field of size 2...

restriction(ring)

Return the restriction of this valuation to ring.

EXAMPLES:

sage: v = QQ.valuation(2)
sage: w = v.restriction(ZZ)
sage: w.domain()
Integer Ring

scale(scalar)

Return this valuation scaled by scalar.

INPUT:
• scalar – a nonnegative rational number or infinity

EXAMPLES:

sage: v = ZZ.valuation(3)
sage: w = v.scale(3)
sage: w(3)
3

Scaling can also be done through multiplication with a scalar:

sage: w/3 == v
True

Multiplication by zero produces the trivial discrete valuation:

sage: w = 0*v
sage: w(3)
0
sage: w(0)
+Infinity

Multiplication by infinity produces the trivial discrete pseudo-valuation:

sage: w = infinity*v
sage: w(3)
+Infinity
sage: w(0)
+Infinity

separating_element(others)
Return an element in the domain of this valuation which has positive valuation with respect to this valu-
ation but negative valuation with respect to the valuations in others.

EXAMPLES:

sage: v2 = QQ.valuation(2)
sage: v3 = QQ.valuation(3)
sage: v5 = QQ.valuation(5)
sage: v2.separating_element([v3,v5])
4/15

shift(x, s)
Shift x in its expansion with respect to uniformizer() by s “digits”.

For nonnegative s, this just returns x multiplied by a power of the uniformizer 𝜋.

For negative s, it does the same but when not over a field, it drops coefficients in the 𝜋-adic expansion
which have negative valuation.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.shift(1, 10)
1024
sage: v.shift(11, -1)
5

For some rings, there is no clear 𝜋-adic expansion. In this case, this method performs negative shifts by
iterated division by the uniformizer and substraction of a lift of the reduction:

sage: R.<x> = ZZ[]
sage: v = ZZ.valuation(2)
sage: w = GaussValuation(R, v)
sage: w.shift(x, 1)
2*x
sage: w.shift(2*x, -1)
x
sage: w.shift(x + 2*x^2, -1)
x^2

simplify(x, error=None, force=False)
Return a simplified version of x.

Produce an element which differs from x by an element of valuation strictly greater than the valuation
of x (or strictly greater than error if set.)

If force is not set, then expensive simplifications may be avoided.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.simplify(6, force=True)
2
sage: v.simplify(6, error=0, force=True)
0

uniformizer()

Return an element in the domain which has positive valuation and generates the value group of this
valuation.

EXAMPLES:

sage: QQ.valuation(11).uniformizer()
11

Trivial valuations have no uniformizer:

sage: from sage.rings.valuation.valuation_space import␣
→˓DiscretePseudoValuationSpace
sage: v = DiscretePseudoValuationSpace(QQ).an_element()
sage: v.is_trivial()
True
sage: v.uniformizer()
Traceback (most recent call last):
...
ValueError: Trivial valuations do not define a uniformizing element

upper_bound(x)

Return an upper bound of this valuation at x.

Use this method to get an approximation of the valuation of x when speed is more important than
accuracy.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.upper_bound(2^10)
10

value_group()

Return the value group of this discrete pseudo-valuation, the discrete additive subgroup of the rational
numbers which is generated by the valuation of the uniformizer().

EXAMPLES:

sage: QQ.valuation(2).value_group()
Additive Abelian Group generated by 1

A pseudo-valuation that is∞ everywhere, does not have a value group:

sage: from sage.rings.valuation.valuation_space import␣
→˓DiscretePseudoValuationSpace
sage: v = DiscretePseudoValuationSpace(QQ).an_element()
sage: v.value_group()
Traceback (most recent call last):
...
ValueError: The trivial pseudo-valuation that is infinity everywhere does␣
→˓not have a value group.

value_semigroup()

Return the value semigroup of this discrete pseudo-valuation, the additive subsemigroup of the rational
numbers which is generated by the valuations of the elements in the domain.

EXAMPLES:

Most commonly, in particular over fields, the semigroup is the group generated by the valuation of the
uniformizer:

sage: G = QQ.valuation(2).value_semigroup(); G
Additive Abelian Semigroup generated by -1, 1
sage: G in AdditiveMagmas().AdditiveAssociative().AdditiveUnital().
→˓AdditiveInverse()
True

If the domain is a discrete valuation ring, then the semigroup consists of the positive elements of the
value_group():

sage: Zp(2).valuation().value_semigroup()
Additive Abelian Semigroup generated by 1

The semigroup can have a more complicated structure when the uniformizer is not in the domain:

sage: v = ZZ.valuation(2)
sage: R.<x> = ZZ[]
sage: w = GaussValuation(R, v)
sage: u = w.augmentation(x, 5/3)
sage: u.value_semigroup()
Additive Abelian Semigroup generated by 1, 5/3

class sage.rings.valuation.valuation_space.ScaleAction

Bases: Action

Action of integers, rationals and the infinity ring on valuations by scaling it.

EXAMPLES:

sage: v = QQ.valuation(5)
sage: from operator import mul
sage: v.parent().get_action(ZZ, mul, self_on_left=False)
Left action by Integer Ring on Discrete pseudo-valuations on Rational Field

5.4 Trivial valuations

AUTHORS:

• Julian Rüth (2016-10-14): initial version

EXAMPLES:

sage: v = valuations.TrivialValuation(QQ); v
Trivial valuation on Rational Field
sage: v(1)
0

class sage.rings.valuation.trivial_valuation.TrivialDiscretePseudoValuation(par-
ent)

Bases: TrivialDiscretePseudoValuation_base, InfiniteDiscretePseudoValuation

The trivial pseudo-valuation that is∞ everywhere.

EXAMPLES:

sage: v = valuations.TrivialPseudoValuation(QQ); v
Trivial pseudo-valuation on Rational Field

lift(X)
Return a lift of X to the domain of this valuation.

EXAMPLES:

sage: v = valuations.TrivialPseudoValuation(QQ)
sage: v.lift(v.residue_ring().zero())
0

reduce(x)
Reduce x modulo the positive elements of this valuation.

EXAMPLES:

sage: v = valuations.TrivialPseudoValuation(QQ)
sage: v.reduce(1)
0

residue_ring()

Return the residue ring of this valuation.

EXAMPLES:

sage: valuations.TrivialPseudoValuation(QQ).residue_ring()
Quotient of Rational Field by the ideal (1)

value_group()

Return the value group of this valuation.

EXAMPLES:

A trivial discrete pseudo-valuation has no value group:

sage: v = valuations.TrivialPseudoValuation(QQ)
sage: v.value_group()
Traceback (most recent call last):
...
ValueError: The trivial pseudo-valuation that is infinity everywhere does not␣
→˓have a value group.

class sage.rings.valuation.trivial_valuation.TrivialDiscretePseudoValuation_base(par-
ent)

Bases: DiscretePseudoValuation

Base class for code shared by trivial valuations.

EXAMPLES:

sage: v = valuations.TrivialPseudoValuation(ZZ); v
Trivial pseudo-valuation on Integer Ring

is_negative_pseudo_valuation()

Return whether this valuation attains the value −∞.

EXAMPLES:

sage: v = valuations.TrivialPseudoValuation(QQ)
sage: v.is_negative_pseudo_valuation()
False

is_trivial()

Return whether this valuation is trivial.

EXAMPLES:

sage: v = valuations.TrivialPseudoValuation(QQ)
sage: v.is_trivial()
True

uniformizer()

Return a uniformizing element for this valuation.

EXAMPLES:

sage: v = valuations.TrivialPseudoValuation(ZZ)
sage: v.uniformizer()
Traceback (most recent call last):
...
ValueError: Trivial valuations do not define a uniformizing element

class sage.rings.valuation.trivial_valuation.TrivialDiscreteValuation(parent)
Bases: TrivialDiscretePseudoValuation_base, DiscreteValuation

The trivial valuation that is zero on nonzero elements.

EXAMPLES:

sage: v = valuations.TrivialValuation(QQ); v
Trivial valuation on Rational Field

extensions(ring)
Return the unique extension of this valuation to ring.

EXAMPLES:

sage: v = valuations.TrivialValuation(ZZ)
sage: v.extensions(QQ)
[Trivial valuation on Rational Field]

lift(X)
Return a lift of X to the domain of this valuation.

EXAMPLES:

sage: v = valuations.TrivialValuation(QQ)
sage: v.lift(v.residue_ring().zero())
0

reduce(x)
Reduce x modulo the positive elements of this valuation.

EXAMPLES:

sage: v = valuations.TrivialValuation(QQ)
sage: v.reduce(1)
1

residue_ring()

Return the residue ring of this valuation.

EXAMPLES:

sage: valuations.TrivialValuation(QQ).residue_ring()
Rational Field

value_group()

Return the value group of this valuation.

EXAMPLES:

A trivial discrete valuation has a trivial value group:

sage: v = valuations.TrivialValuation(QQ)
sage: v.value_group()
Trivial Additive Abelian Group

class sage.rings.valuation.trivial_valuation.TrivialValuationFactory(clazz, parent,
*args,
**kwargs)

Bases: UniqueFactory

Create a trivial valuation on domain.

EXAMPLES:

sage: v = valuations.TrivialValuation(QQ); v
Trivial valuation on Rational Field
sage: v(1)
0

create_key(domain)
Create a key that identifies this valuation.

EXAMPLES:

sage: valuations.TrivialValuation(QQ) is valuations.TrivialValuation(QQ) #␣
→˓indirect doctest
True

create_object(version, key, **extra_args)
Create a trivial valuation from key.

EXAMPLES:

sage: valuations.TrivialValuation(QQ) # indirect doctest
Trivial valuation on Rational Field

5.5 Gauss valuations on polynomial rings

This file implements Gauss valuations for polynomial rings, i.e. discrete valuations which assign to a polynomial the
minimal valuation of its coefficients.

AUTHORS:

• Julian Rüth (2013-04-15): initial version

EXAMPLES:

A Gauss valuation maps a polynomial to the minimal valuation of any of its coefficients:

sage: R.<x> = QQ[]
sage: v0 = QQ.valuation(2)
sage: v = GaussValuation(R, v0); v
Gauss valuation induced by 2-adic valuation
sage: v(2*x + 2)
1

Gauss valuations can also be defined iteratively based on valuations over polynomial rings:

sage: v = v.augmentation(x, 1/4); v
[ Gauss valuation induced by 2-adic valuation, v(x) = 1/4 ]
sage: v = v.augmentation(x^4+2*x^3+2*x^2+2*x+2, 4/3); v
[ Gauss valuation induced by 2-adic valuation, v(x) = 1/4, v(x^4 + 2*x^3 + 2*x^2 +␣
→˓2*x + 2) = 4/3 ]
sage: S.<T> = R[]
sage: w = GaussValuation(S, v); w
Gauss valuation induced by [ Gauss valuation induced by 2-adic valuation, v(x) = 1/4,␣
→˓v(x^4 + 2*x^3 + 2*x^2 + 2*x + 2) = 4/3 ]
sage: w(2*T + 1)
0

class sage.rings.valuation.gauss_valuation.GaussValuationFactory

Bases: UniqueFactory

Create a Gauss valuation on domain.

INPUT:

• domain – a univariate polynomial ring

• v – a valuation on the base ring of domain, the underlying valuation on the constants of the polynomial ring
(if unspecified take the natural valuation on the valued ring domain.)

EXAMPLES:

The Gauss valuation is the minimum of the valuation of the coefficients:

sage: v = QQ.valuation(2)
sage: R.<x> = QQ[]
sage: w = GaussValuation(R, v)
sage: w(2)
1
sage: w(x)
0
sage: w(x + 2)
0

create_key(domain, v=None)
Normalize and check the parameters to create a Gauss valuation.

create_object(version, key, **extra_args)
Create a Gauss valuation from normalized parameters.

class sage.rings.valuation.gauss_valuation.GaussValuation_generic(parent, v)
Bases: NonFinalInductiveValuation

A Gauss valuation on a polynomial ring domain.

INPUT:

• domain – a univariate polynomial ring over a valued ring 𝑅

• v – a discrete valuation on 𝑅

EXAMPLES:

sage: R = Zp(3,5)
sage: S.<x> = R[] #␣
→˓needs sage.libs.ntl
sage: v0 = R.valuation()
sage: v = GaussValuation(S, v0); v #␣
→˓needs sage.libs.ntl
Gauss valuation induced by 3-adic valuation

sage: S.<x> = QQ[]
sage: v = GaussValuation(S, QQ.valuation(5)); v
Gauss valuation induced by 5-adic valuation

E()

Return the ramification index of this valuation over its underlying Gauss valuation, i.e., 1.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.E()
1

F()

Return the degree of the residue field extension of this valuation over the Gauss valuation, i.e., 1.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.F()
1

augmentation_chain()

Return a list with the chain of augmentations down to the underlying Gauss valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.augmentation_chain()
[Gauss valuation induced by 2-adic valuation]

change_domain(ring)
Return this valuation as a valuation over ring.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: R.<x> = ZZ[]
sage: w = GaussValuation(R, v)
sage: w.change_domain(QQ[�x�])
Gauss valuation induced by 2-adic valuation

element_with_valuation(s)
Return a polynomial of minimal degree with valuation s.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: v.element_with_valuation(-2)
1/4

equivalence_unit(s, reciprocal=False)
Return an equivalence unit of valuation s.

INPUT:

• s – an element of the value_group()

• reciprocal – boolean (default: False); whether or not to return the equivalence unit as theequiv-
alence_reciprocal() of the equivalence unit of valuation -s

EXAMPLES:

sage: # needs sage.libs.ntl
sage: S.<x> = Qp(3,5)[]
sage: v = GaussValuation(S)
sage: v.equivalence_unit(2)
3^2 + O(3^7)
sage: v.equivalence_unit(-2)
3^-2 + O(3^3)

extensions(ring)
Return the extensions of this valuation to ring.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: R.<x> = ZZ[]
sage: w = GaussValuation(R, v)
sage: w.extensions(GaussianIntegers()[�x�]) #␣
→˓needs sage.rings.number_field
[Gauss valuation induced by 2-adic valuation]

is_gauss_valuation()

Return whether this valuation is a Gauss valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.is_gauss_valuation()
True

is_trivial()

Return whether this is a trivial valuation (sending everything but zero to zero.)

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ))
sage: v.is_trivial()
True

lift(F)
Return a lift of F.

INPUT:

• F – a polynomial over the residue_ring() of this valuation

OUTPUT:

a (possibly non-monic) polynomial in the domain of this valuation which reduces to F

EXAMPLES:

sage: # needs sage.libs.ntl
sage: S.<x> = Qp(3,5)[]
sage: v = GaussValuation(S)
sage: f = x^2 + 2*x + 16
sage: F = v.reduce(f); F
x^2 + 2*x + 1
sage: g = v.lift(F); g
(1 + O(3^5))*x^2 + (2 + O(3^5))*x + 1 + O(3^5)
sage: v.is_equivalent(f,g)
True
sage: g.parent() is v.domain()
True

See also

reduce()

lift_to_key(F)
Lift the irreducible polynomial F from the residue_ring() to a key polynomial over this valuation.

INPUT:

• F – an irreducible non-constant monic polynomial in residue_ring() of this valuation

OUTPUT:

A polynomial 𝑓 in the domain of this valuation which is a key polynomial for this valuation and which, for a
suitable equivalence unit 𝑅, satisfies that the reduction of 𝑅𝑓 is F

EXAMPLES:

sage: R.<u> = QQ
sage: S.<x> = R[]
sage: v = GaussValuation(S, QQ.valuation(2))
sage: y = v.residue_ring().gen()
sage: f = v.lift_to_key(y^2 + y + 1); f
x^2 + x + 1

lower_bound(f)
Return a lower bound of this valuation at f.

Use this method to get an approximation of the valuation of f when speed is more important than accuracy.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.lower_bound(1024*x + 2)
1
sage: v(1024*x + 2)
1

monic_integral_model(G)
Return a monic integral irreducible polynomial which defines the same extension of the base ring of the
domain as the irreducible polynomial G together with maps between the old and the new polynomial.

EXAMPLES:

sage: R.<x> = Qp(2, 5)[] #␣
→˓needs sage.libs.ntl
sage: v = GaussValuation(R) #␣
→˓needs sage.libs.ntl
sage: v.monic_integral_model(5*x^2 + 1/2*x + 1/4) #␣
→˓needs sage.libs.ntl
(Ring endomorphism of Univariate Polynomial Ring in x over 2-adic Field with␣
→˓capped relative precision 5

Defn: (1 + O(2^5))*x |--> (2^-1 + O(2^4))*x,
Ring endomorphism of Univariate Polynomial Ring in x over 2-adic Field with␣
→˓capped relative precision 5

Defn: (1 + O(2^5))*x |--> (2 + O(2^6))*x,
(1 + O(2^5))*x^2 + (1 + 2^2 + 2^3 + O(2^5))*x + 1 + 2^2 + 2^3 + O(2^5))

reduce(f , check=True, degree_bound=None, coefficients=None, valuations=None)
Return the reduction of f modulo this valuation.

INPUT:

• f – an integral element of the domain of this valuation

• check – whether or not to check whether f has nonnegative valuation (default: True)

• degree_bound – an a-priori known bound on the degree of the result which can speed up the com-
putation (default: not set)

• coefficients – the coefficients off as produced bycoefficients() orNone (default: None);
ignored

• valuations – the valuations of coefficients or None (default: None); ignored

OUTPUT: a polynomial in the residue_ring() of this valuation

EXAMPLES:

sage: # needs sage.libs.ntl
sage: S.<x> = Qp(2,5)[]
sage: v = GaussValuation(S)
sage: f = x^2 + 2*x + 16

(continues on next page)

(continued from previous page)

sage: v.reduce(f)
x^2
sage: v.reduce(f).parent() is v.residue_ring()
True

The reduction is only defined for integral elements:

sage: f = x^2/2 #␣
→˓needs sage.libs.ntl
sage: v.reduce(f) #␣
→˓needs sage.libs.ntl
Traceback (most recent call last):
...
ValueError: reduction not defined for non-integral elements and (2^-1 + O(2^
→˓4))*x^2 is not integral over Gauss valuation induced by 2-adic valuation

See also

lift()

residue_ring()

Return the residue ring of this valuation, i.e., the elements of valuation zero module the elements of positive
valuation.

EXAMPLES:

sage: S.<x> = Qp(2,5)[] #␣
→˓needs sage.libs.ntl
sage: v = GaussValuation(S) #␣
→˓needs sage.libs.ntl
sage: v.residue_ring() #␣
→˓needs sage.libs.ntl
Univariate Polynomial Ring in x over Finite Field of size 2 (using ...)

restriction(ring)
Return the restriction of this valuation to ring.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: R.<x> = ZZ[]
sage: w = GaussValuation(R, v)
sage: w.restriction(ZZ)
2-adic valuation

scale(scalar)
Return this valuation scaled by scalar.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: 3*v # indirect doctest
Gauss valuation induced by 3 * 2-adic valuation

simplify(f , error=None, force=False, size_heuristic_bound=32, effective_degree=None, phiadic=True)
Return a simplified version of f.

Produce an element which differs from f by an element of valuation strictly greater than the valuation of f
(or strictly greater than error if set.)

INPUT:

• f – an element in the domain of this valuation

• error – a rational, infinity, or None (default: None), the error allowed to introduce through the
simplification

• force – whether or not to simplify f even if there is heuristically no change in the coefficient size of f
expected (default: False)

• effective_degree – when set, assume that coefficients beyond effective_degree can be
safely dropped (default: None)

• size_heuristic_bound – when force is not set, the expected factor by which the coefficients
need to shrink to perform an actual simplification (default: 32)

• phiadic – whether to simplify in the 𝑥-adic expansion; the parameter is ignored as no other simpli-
fication is implemented

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: f = x^10/2 + 1
sage: v.simplify(f)
(2^-1 + O(2^4))*x^10 + 1 + O(2^5)

uniformizer()

Return a uniformizer of this valuation, i.e., a uniformizer of the valuation of the base ring.

EXAMPLES:

sage: S.<x> = QQ[]
sage: v = GaussValuation(S, QQ.valuation(5))
sage: v.uniformizer()
5
sage: v.uniformizer().parent() is S
True

upper_bound(f)

Return an upper bound of this valuation at f.

Use this method to get an approximation of the valuation of f when speed is more important than accuracy.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.upper_bound(1024*x + 1)
10

(continues on next page)

(continued from previous page)

sage: v(1024*x + 1)
0

valuations(f , coefficients=None, call_error=False)
Return the valuations of the 𝑓𝑖𝜑𝑖 in the expansion 𝑓 =

∑︀
𝑓𝑖𝜑

𝑖.

INPUT:

• f – a polynomial in the domain of this valuation

• coefficients – the coefficients off as produced bycoefficients() orNone (default: None);
this can be used to speed up the computation when the expansion of f is already known from a previous
computation.

• call_error – whether or not to speed up the computation by assuming that the result is only used to
compute the valuation of f (default: False)

OUTPUT: list, each entry a rational numbers or infinity, the valuations of 𝑓0, 𝑓1𝜑, . . .

EXAMPLES:

sage: R = ZZ
sage: S.<x> = R[]
sage: v = GaussValuation(S, R.valuation(2))
sage: f = x^2 + 2*x + 16
sage: list(v.valuations(f))
[4, 1, 0]

value_group()

Return the value group of this valuation.

EXAMPLES:

sage: S.<x> = QQ[]
sage: v = GaussValuation(S, QQ.valuation(5))
sage: v.value_group()
Additive Abelian Group generated by 1

value_semigroup()

Return the value semigroup of this valuation.

EXAMPLES:

sage: S.<x> = QQ[]
sage: v = GaussValuation(S, QQ.valuation(5))
sage: v.value_semigroup()
Additive Abelian Semigroup generated by -1, 1

5.6 Valuations on polynomial rings based on 𝜑-adic expansions

This file implements a base class for discrete valuations on polynomial rings, defined by a 𝜑-adic expansion.

AUTHORS:

• Julian Rüth (2013-04-15): initial version

EXAMPLES:

The Gauss valuation is a simple example of a valuation that relies on 𝜑-adic expansions:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))

In this case, 𝜑 = 𝑥, so the expansion simply lists the coefficients of the polynomial:

sage: f = x^2 + 2*x + 2
sage: list(v.coefficients(f))
[2, 2, 1]

Often only the first few coefficients are necessary in computations, so for performance reasons, coefficients are computed
lazily:

sage: v.coefficients(f)
<generator object ...coefficients at 0x...>

Another example of a DevelopingValuation is an augmented valuation:

sage: w = v.augmentation(x^2 + x + 1, 3)

Here, the expansion lists the remainders of repeated division by 𝑥2 + 𝑥+ 1:

sage: list(w.coefficients(f))
[x + 1, 1]

class sage.rings.valuation.developing_valuation.DevelopingValuation(parent, phi)
Bases: DiscretePseudoValuation

Abstract base class for a discrete valuation of polynomials defined over the polynomial ring domain by the 𝜑-adic
development.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(7))

coefficients(f)
Return the 𝜑-adic expansion of f.

INPUT:

• f – a monic polynomial in the domain of this valuation

OUTPUT:

An iterator 𝑓0, 𝑓1, . . . , 𝑓𝑛 of polynomials in the domain of this valuation such that 𝑓 =
∑︀

𝑖 𝑓𝑖𝜑
𝑖

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R = Qp(2,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: f = x^2 + 2*x + 3
sage: list(v.coefficients(f)) # note that these constants are in the␣
→˓polynomial ring
[1 + 2 + O(2^5), 2 + O(2^6), 1 + O(2^5)]
sage: v = v.augmentation( x^2 + x + 1, 1)
sage: list(v.coefficients(f))
[(1 + O(2^5))*x + 2 + O(2^5), 1 + O(2^5)]

effective_degree(f , valuations=None)
Return the effective degree of f with respect to this valuation.

The effective degree of 𝑓 is the largest 𝑖 such that the valuation of 𝑓 and the valuation of 𝑓𝑖𝜑𝑖 in the de-
velopment 𝑓 =

∑︀
𝑗 𝑓𝑗𝜑

𝑗 coincide (see [Mac1936II] p.497.)

INPUT:

• f – a nonzero polynomial in the domain of this valuation

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R = Zp(2,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.effective_degree(x)
1
sage: v.effective_degree(2*x + 1)
0

newton_polygon(f , valuations=None)
Return the Newton polygon of the 𝜑-adic development of f.

INPUT:

• f – a polynomial in the domain of this valuation

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R = Qp(2,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: f = x^2 + 2*x + 3
sage: v.newton_polygon(f) #␣
→˓needs sage.geometry.polyhedron
Finite Newton polygon with 2 vertices: (0, 0), (2, 0)
sage: v = v.augmentation( x^2 + x + 1, 1)
sage: v.newton_polygon(f) #␣
→˓needs sage.geometry.polyhedron
Finite Newton polygon with 2 vertices: (0, 0), (1, 1)
sage: v.newton_polygon( f * v.phi()^3 ) #␣
→˓needs sage.geometry.polyhedron
Finite Newton polygon with 2 vertices: (3, 3), (4, 4)

phi()

Return the polynomial 𝜑, the key polynomial of this valuation.

EXAMPLES:

sage: R = Zp(2,5)
sage: S.<x> = R[] #␣
→˓needs sage.libs.ntl
sage: v = GaussValuation(S) #␣
→˓needs sage.libs.ntl
sage: v.phi() #␣
→˓needs sage.libs.ntl
(1 + O(2^5))*x

valuations(f)
Return the valuations of the 𝑓𝑖𝜑𝑖 in the expansion 𝑓 =

∑︀
𝑓𝑖𝜑

𝑖.

INPUT:

• f – a polynomial in the domain of this valuation

OUTPUT:

A list, each entry a rational numbers or infinity, the valuations of 𝑓0, 𝑓1𝜑, . . .

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R = Qp(2,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S, R.valuation())
sage: f = x^2 + 2*x + 16
sage: list(v.valuations(f))
[4, 1, 0]

5.7 Inductive valuations on polynomial rings

This module provides functionality for inductive valuations, i.e., finite chains of augmented valuations on top of
a Gauss valuation.

AUTHORS:

• Julian Rüth (2016-11-01): initial version

EXAMPLES:

A Gauss valuation is an example of an inductive valuation:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))

Generally, an inductive valuation is an augmentation of an inductive valuation, i.e., a valuation that was created from a
Gauss valuation in a finite number of augmentation steps:

sage: w = v.augmentation(x, 1)
sage: w = w.augmentation(x^2 + 2*x + 4, 3)

REFERENCES:

Inductive valuations are originally discussed in [Mac1936I] and [Mac1936II]. An introduction is also given in Chapter 4
of [Rüt2014].

class sage.rings.valuation.inductive_valuation.FinalInductiveValuation(parent,
phi)

Bases: InductiveValuation

Abstract base class for an inductive valuation which cannot be augmented further.

class sage.rings.valuation.inductive_valuation.FiniteInductiveValuation(parent,
phi)

Bases: InductiveValuation, DiscreteValuation

Abstract base class for iterated augmented valuations on top of a Gauss valuationwhich is a discrete
valuation, i.e., the last key polynomial has finite valuation.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ))

extensions(other)
Return the extensions of this valuation to other.

EXAMPLES:

sage: R.<x> = ZZ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(ZZ))
sage: K.<x> = FunctionField(QQ)
sage: v.extensions(K)
[Trivial valuation on Rational Field]

class sage.rings.valuation.inductive_valuation.InductiveValuation(parent, phi)
Bases: DevelopingValuation

Abstract base class for iterated augmented valuations on top of a Gauss valuation.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(5))

E()

Return the ramification index of this valuation over its underlying Gauss valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.E()
1

F()

Return the residual degree of this valuation over its Gauss extension.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.F()
1

augmentation_chain()

Return a list with the chain of augmentations down to the underlying Gauss valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.augmentation_chain()
[Gauss valuation induced by 2-adic valuation]

element_with_valuation(s)
Return a polynomial of minimal degree with valuation s.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: v.element_with_valuation(-2)
1/4

Depending on the base ring, an element of valuation s might not exist:

sage: R.<x> = ZZ[]
sage: v = GaussValuation(R, ZZ.valuation(2))
sage: v.element_with_valuation(-2)
Traceback (most recent call last):
...
ValueError: s must be in the value semigroup of this valuation
but -2 is not in Additive Abelian Semigroup generated by 1

equivalence_reciprocal(f , coefficients=None, valuations=None, check=True)
Return an equivalence reciprocal of f.

An equivalence reciprocal of 𝑓 is a polynomial ℎ such that 𝑓 · ℎ is equivalent to 1 modulo this valuation (see
[Mac1936II] p.497.)

INPUT:

• f – a polynomial in the domain of this valuation which is an equivalence_unit()

• coefficients – the coefficients of f in the phi()-adic expansion if known (default: None)

• valuations – the valuations of coefficients if known (default: None)

• check – whether or not to check the validity of f (default: True)

Warning

This method may not work over 𝑝-adic rings due to problems with the xgcd implementation there.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R = Zp(3,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: f = 3*x + 2
sage: h = v.equivalence_reciprocal(f); h
2 + O(3^5)
sage: v.is_equivalent(f*h, 1)
True

In an extended valuation over an extension field:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v = v.augmentation(x^2 + x + u, 1)
sage: f = 2*x + u
sage: h = v.equivalence_reciprocal(f); h
(u + 1) + O(2^5)
sage: v.is_equivalent(f*h, 1)
True

Extending the valuation once more:

sage: # needs sage.libs.ntl
sage: v = v.augmentation((x^2 + x + u)^2 + 2*x*(x^2 + x + u) + 4*x, 3)
sage: h = v.equivalence_reciprocal(f); h
(u + 1) + O(2^5)
sage: v.is_equivalent(f*h, 1)
True

equivalence_unit(s, reciprocal=False)
Return an equivalence unit of valuation s.

INPUT:

• s – an element of the value_group()

• reciprocal – boolean (default: False); whether or not to return the equivalence unit as theequiv-
alence_reciprocal() of the equivalence unit of valuation -s.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: S.<x> = Qp(3,5)[]
sage: v = GaussValuation(S)
sage: v.equivalence_unit(2)
3^2 + O(3^7)
sage: v.equivalence_unit(-2)
3^-2 + O(3^3)

Note that this might fail for negative s if the domain is not defined over a field:

sage: v = ZZ.valuation(2)
sage: R.<x> = ZZ[]
sage: w = GaussValuation(R, v)

(continues on next page)

(continued from previous page)

sage: w.equivalence_unit(1)
2
sage: w.equivalence_unit(-1)
Traceback (most recent call last):
...
ValueError: s must be in the value semigroup of this valuation
but -1 is not in Additive Abelian Semigroup generated by 1

is_equivalence_unit(f , valuations=None)
Return whether the polynomial f is an equivalence unit, i.e., an element of effective_degree() zero
(see [Mac1936II] p.497.)

INPUT:

• f – a polynomial in the domain of this valuation

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R = Zp(2,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.is_equivalence_unit(x)
False
sage: v.is_equivalence_unit(S.zero())
False
sage: v.is_equivalence_unit(2*x + 1)
True

is_gauss_valuation()

Return whether this valuation is a Gauss valuation over the domain.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.is_gauss_valuation()
True

monic_integral_model(G)
Return a monic integral irreducible polynomial which defines the same extension of the base ring of the
domain as the irreducible polynomial G together with maps between the old and the new polynomial.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: v.monic_integral_model(5*x^2 + 1/2*x + 1/4)
(Ring endomorphism of Univariate Polynomial Ring in x over Rational Field

Defn: x |--> 1/2*x,
Ring endomorphism of Univariate Polynomial Ring in x over Rational Field

Defn: x |--> 2*x,
x^2 + 1/5*x + 1/5)

mu()

Return the valuation of phi().

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: v.mu()
0

class sage.rings.valuation.inductive_valuation.InfiniteInductiveValuation(par-
ent,
base_val-
ua-
tion)

Bases: FinalInductiveValuation, InfiniteDiscretePseudoValuation

Abstract base class for an inductive valuation which is not discrete, i.e., which assigns infinite valuation to its last
key polynomial.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, infinity)

change_domain(ring)
Return this valuation over ring.

EXAMPLES:

We can turn an infinite valuation into a valuation on the quotient:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, infinity)
sage: w.change_domain(R.quo(x^2 + x + 1))
2-adic valuation

class sage.rings.valuation.inductive_valuation.NonFinalInductiveValuation(par-
ent,
phi)

Bases: FiniteInductiveValuation, DiscreteValuation

Abstract base class for iterated augmented valuations on top of a Gauss valuation which can be
extended further through augmentation().

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v = v.augmentation(x^2 + x + u, 1)

augmentation(phi, mu, check=True)
Return the inductive valuation which extends this valuation by mapping phi to mu.

INPUT:

• phi – a polynomial in the domain of this valuation; this must be a key polynomial, see is_key() for
properties of key polynomials.

• mu – a rational number or infinity, the valuation of phi in the extended valuation

• check – boolean (default: True); whether or not to check the correctness of the parameters

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v = v.augmentation(x^2 + x + u, 1)
sage: v = v.augmentation((x^2 + x + u)^2 + 2*x*(x^2 + x + u) + 4*x, 3)
sage: v
[ Gauss valuation induced by 2-adic valuation,

v((1 + O(2^5))*x^2 + (1 + O(2^5))*x + u + O(2^5)) = 1,
v((1 + O(2^5))*x^4
+ (2^2 + O(2^6))*x^3
+ (1 + (u + 1)*2 + O(2^5))*x^2
+ ((u + 1)*2^2 + O(2^6))*x
+ (u + 1) + (u + 1)*2 + (u + 1)*2^2 + (u + 1)*2^3 + (u + 1)*2^4 + O(2^5))␣

→˓= 3 ]

See also

augmented_valuation

equivalence_decomposition(f , assume_not_equivalence_unit=False, coefficients=None,
valuations=None, compute_unit=True, degree_bound=None)

Return an equivalence decomposition of f, i.e., a polynomial 𝑔(𝑥) = 𝑒(𝑥)
∏︀

𝑖 𝜑𝑖(𝑥) with 𝑒(𝑥) an equiv-
alence unit and the 𝜑𝑖 key polynomials such that f is_equivalent() to 𝑔.

INPUT:

• f – a nonzero polynomial in the domain of this valuation

• assume_not_equivalence_unit – whether or not to assume that f is not an equivalence
unit (default: False)

• coefficients – the coefficients of f in the phi()-adic expansion if known (default: None)

• valuations – the valuations of coefficients if known (default: None)

• compute_unit – whether or not to compute the unit part of the decomposition (default: True)

• degree_bound – a bound on the degree of the _equivalence_reduction() of f (default:
None)

ALGORITHM:

We use the algorithm described in Theorem 4.4 of [Mac1936II]. After removing all factors 𝜑 from a poly-
nomial 𝑓 , there is an equivalence unit 𝑅 such that 𝑅𝑓 has valuation zero. Now 𝑅𝑓 can be factored as

∏︀
𝑖 𝛼𝑖

over the residue_field(). Lifting all 𝛼𝑖 to key polynomials 𝜑𝑖 gives𝑅𝑓 =
∏︀

𝑖 𝑅𝑖𝑓𝑖 for suitable equiv-
alence units 𝑅𝑖 (see lift_to_key()). Taking 𝑅′ an equivalence_reciprocal() of 𝑅, we have
𝑓 equivalent to (𝑅′ ∏︀

𝑖 𝑅𝑖)
∏︀

𝑖 𝜑𝑖.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,10)

(continues on next page)

sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.equivalence_decomposition(S.zero())
Traceback (most recent call last):
...
ValueError: equivalence decomposition of zero is not defined
sage: v.equivalence_decomposition(S.one())
1 + O(2^10)
sage: v.equivalence_decomposition(x^2+2)
((1 + O(2^10))*x)^2
sage: v.equivalence_decomposition(x^2+1)
((1 + O(2^10))*x + 1 + O(2^10))^2

A polynomial that is an equivalence unit, is returned as the unit part of a Factorization, leading to a
unit non-minimal degree:

sage: w = v.augmentation(x, 1) #␣
→˓needs sage.libs.ntl
sage: F = w.equivalence_decomposition(x^2+1); F #␣
→˓needs sage.libs.ntl
(1 + O(2^10))*x^2 + 1 + O(2^10)
sage: F.unit() #␣
→˓needs sage.libs.ntl
(1 + O(2^10))*x^2 + 1 + O(2^10)

However, if the polynomial has a non-unit factor, then the unit might be replaced by a factor of lower degree:

sage: f = x * (x^2 + 1) #␣
→˓needs sage.libs.ntl
sage: F = w.equivalence_decomposition(f); F #␣
→˓needs sage.libs.ntl
(1 + O(2^10))*x
sage: F.unit() #␣
→˓needs sage.libs.ntl
1 + O(2^10)

Examples over an iterated unramified extension:

sage: # needs sage.libs.ntl
sage: v = v.augmentation(x^2 + x + u, 1)
sage: v = v.augmentation((x^2 + x + u)^2 + 2*x*(x^2 + x + u) + 4*x, 3)
sage: v.equivalence_decomposition(x)
(1 + O(2^10))*x
sage: F = v.equivalence_decomposition( v.phi() )
sage: len(F)
1
sage: F = v.equivalence_decomposition( v.phi() * (x^4 + 4*x^3 + (7 + 2*u)*x^2␣
→˓+ (8 + 4*u)*x + 1023 + 3*u) )
sage: len(F)
2

is_equivalence_irreducible(f , coefficients=None, valuations=None)
Return whether the polynomial f is equivalence-irreducible, i.e., whether its equivalence_decompo-
sition() is trivial.

ALGORITHM:

We use the same algorithm as in equivalence_decomposition() we just do not lift the result to key
polynomials.

INPUT:

• f – a non-constant polynomial in the domain of this valuation

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.is_equivalence_irreducible(x)
True
sage: v.is_equivalence_irreducible(x^2)
False
sage: v.is_equivalence_irreducible(x^2 + 2)
False

is_key(phi, explain=False, assume_equivalence_irreducible=False)
Return whether phi is a key polynomial for this valuation, i.e., whether it is monic, whether it is_equiv-
alence_irreducible(), and whether it is is_minimal().

INPUT:

• phi – a polynomial in the domain of this valuation

• explain – boolean (default: False); if True, return a string explaining why phi is not a key poly-
nomial

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.is_key(x)
True
sage: v.is_key(2*x, explain=True)
(False, �phi must be monic�)
sage: v.is_key(x^2, explain=True)
(False, �phi must be equivalence irreducible�)
sage: w = v.augmentation(x, 1)
sage: w.is_key(x + 1, explain = True)
(False, �phi must be minimal�)

is_minimal(f , assume_equivalence_irreducible=False)
Return whether the polynomial f is minimal with respect to this valuation.

A polynomial 𝑓 is minimal with respect to 𝑣 if it is not a constant and any nonzero polynomial ℎ which is
𝑣-divisible by 𝑓 has at least the degree of 𝑓 .

A polynomial ℎ is 𝑣-divisible by 𝑓 if there is a polynomial 𝑐 such that 𝑓𝑐 is_equivalent() to ℎ.

ALGORITHM:

Based on Theorem 9.4 of [Mac1936II].

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.is_minimal(x + 1)
True
sage: w = v.augmentation(x, 1)
sage: w.is_minimal(x + 1)
False

lift_to_key(F)
Lift the irreducible polynomial F from the residue_ring() to a key polynomial over this valuation.

INPUT:

• F – an irreducible non-constant monic polynomial in residue_ring() of this valuation

OUTPUT:

A polynomial 𝑓 in the domain of this valuation which is a key polynomial for this valuation and which is such
that an augmentation() with this polynomial adjoins a root of F to the resulting residue_ring().

More specifically, if F is not the generator of the residue ring, then multiplying f with the equiva-
lence_reciprocal() of the equivalence_unit() of the valuation of f, produces a unit which
reduces to F.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,10)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: y = v.residue_ring().gen()
sage: u0 = v.residue_ring().base_ring().gen()
sage: f = v.lift_to_key(y^2 + y + u0); f
(1 + O(2^10))*x^2 + (1 + O(2^10))*x + u + O(2^10)

mac_lane_step(G, principal_part_bound=None, assume_squarefree=False,
assume_equivalence_irreducible=False, report_degree_bounds_and_caches=False,
coefficients=None, valuations=None, check=True, allow_equivalent_key=True)

Perform an approximation step towards the squarefree monic non-constant integral polynomial G which is not
an equivalence unit.

This performs the individual steps that are used in mac_lane_approximants().

INPUT:

• G – a squarefree monic non-constant integral polynomial G which is not an equivalence unit

• principal_part_bound – integer or None (default: None), a bound on the length of the principal
part, i.e., the section of negative slope, of the Newton polygon of G

• assume_squarefree – whether or not to assume that G is squarefree (default: False)

• assume_equivalence_irreducible – whether or not to assume that G is equivalence irre-
ducible (default: False)

• report_degree_bounds_and_caches – whether or not to include internal state with the re-
turned value (used by mac_lane_approximants() to speed up sequential calls)

• coefficients – the coefficients of G in the phi()-adic expansion if known (default: None)

• valuations – the valuations of coefficients if known (default: None)

• check – whether to check that G is a squarefree monic non-constant integral polynomial and not an
equivalence unit (default: True)

• allow_equivalent_key – whether to return valuations which end in essentially the same key poly-
nomial as this valuation but have a higher valuation assigned to that key polynomial (default: True)

EXAMPLES:

We can use this method to perform the individual steps of mac_lane_approximants():

sage: R.<x> = QQ[]
sage: v = QQ.valuation(2)
sage: f = x^36 + 1160/81*x^31 + 9920/27*x^30 + 1040/81*x^26 + 52480/81*x^25 +␣
→˓220160/81*x^24 - 5120/81*x^21 - 143360/81*x^20 - 573440/81*x^19 + 12451840/
→˓81*x^18 - 266240/567*x^16 - 20316160/567*x^15 - 198737920/189*x^14 -␣
→˓1129840640/81*x^13 - 1907359744/27*x^12 + 8192/81*x^11 + 655360/81*x^10 +␣
→˓5242880/21*x^9 + 2118123520/567*x^8 + 15460204544/567*x^7 + 6509559808/81*x^
→˓6 - 16777216/567*x^2 - 268435456/567*x - 1073741824/567
sage: v.mac_lane_approximants(f) #␣
→˓needs sage.geometry.polyhedron
[[ Gauss valuation induced by 2-adic valuation, v(x + 2056) = 23/2 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 11/9 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 2/5, v(x^5 + 4) = 7/2␣
→˓],
[ Gauss valuation induced by 2-adic valuation, v(x) = 3/5, v(x^10 + 8*x^5 +␣
→˓64) = 7 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 3/5, v(x^5 + 8) = 5 ]]

Starting from the Gauss valuation, aMacLane step branches off with some linear key polynomials in the above
example:

sage: v0 = GaussValuation(R, v)
sage: V1 = sorted(v0.mac_lane_step(f)); V1 #␣
→˓needs sage.geometry.polyhedron
[[ Gauss valuation induced by 2-adic valuation, v(x) = 2/5 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 3/5 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 11/9 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 3 ]]

The computation of MacLane approximants would now perform a MacLane step on each of these branches,
note however, that a direct call to this method might produce some unexpected results:

sage: V1[1].mac_lane_step(f) #␣
→˓needs sage.geometry.polyhedron
[[ Gauss valuation induced by 2-adic valuation, v(x) = 3/5, v(x^5 + 8) = 5 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 3/5, v(x^10 + 8*x^5 +␣
→˓64) = 7 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 3 ],
[ Gauss valuation induced by 2-adic valuation, v(x) = 11/9 ]]

Note how this detected the two augmentations of V1[1] but also two other valuations that we had seen
in the previous step and that are greater than V1[1]. To ignore such trivial augmentations, we can set
allow_equivalent_key:

sage: V1[1].mac_lane_step(f, allow_equivalent_key=False) #␣
→˓needs sage.geometry.polyhedron
[[ Gauss valuation induced by 2-adic valuation, v(x) = 3/5, v(x^5 + 8) = 5 ],

(continues on next page)

[ Gauss valuation induced by 2-adic valuation, v(x) = 3/5, v(x^10 + 8*x^5 +␣
→˓64) = 7 ]]

minimal_representative(f)
Return a minimal representative for f, i.e., a pair 𝑒, 𝑎 such that f is_equivalent() to 𝑒𝑎, 𝑒 is an
equivalence unit, and 𝑎 is_minimal() and monic.

INPUT:

• f – a nonzero polynomial which is not an equivalence unit

OUTPUT: a factorization which has 𝑒 as its unit and 𝑎 as its unique factor

ALGORITHM:

We use the algorithm described in the proof of Lemma 4.1 of [Mac1936II]. In the expansion 𝑓 =
∑︀

𝑖 𝑓𝑖𝜑
𝑖

take 𝑒 = 𝑓𝑖 for the largest 𝑖 with 𝑓𝑖𝜑
𝑖 minimal (see effective_degree()). Let ℎ be the equiv-

alence_reciprocal() of 𝑒 and take 𝑎 given by the terms of minimal valuation in the expansion of
𝑒𝑓 .

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4,10)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: v.minimal_representative(x + 2)
(1 + O(2^10))*x

sage: # needs sage.libs.ntl
sage: v = v.augmentation(x, 1)
sage: v.minimal_representative(x + 2)
(1 + O(2^10))*x + 2 + O(2^11)
sage: f = x^3 + 6*x + 4
sage: F = v.minimal_representative(f); F
(2 + 2^2 + O(2^11)) * ((1 + O(2^10))*x + 2 + O(2^11))
sage: v.is_minimal(F[0][0])
True
sage: v.is_equivalent(F.prod(), f)
True

5.8 Augmented valuations on polynomial rings

Implements augmentations of (inductive) valuations.

AUTHORS:

• Julian Rüth (2013-04-15): initial version

EXAMPLES:

Starting from a Gauss valuation, we can create augmented valuations on polynomial rings:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x, 1); w
[ Gauss valuation induced by 2-adic valuation, v(x) = 1 ]

(continues on next page)

sage: w(x)
1

This also works for polynomial rings over base rings which are not fields. However, much of the functionality is only
available over fields:

sage: R.<x> = ZZ[]
sage: v = GaussValuation(R, ZZ.valuation(2))
sage: w = v.augmentation(x, 1); w
[ Gauss valuation induced by 2-adic valuation, v(x) = 1 ]
sage: w(x)
1

REFERENCES:

Augmentations are described originally in [Mac1936I] and [Mac1936II]. An overview can also be found in Chapter 4 of
[Rüt2014].

class sage.rings.valuation.augmented_valuation.AugmentedValuationFactory

Bases: UniqueFactory

Factory for augmented valuations.

EXAMPLES:

This factory is not meant to be called directly. Instead, augmentation() of a valuation should be called:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x, 1) # indirect doctest

Note that trivial parts of the augmented valuation might be dropped, so you should not rely on _base_valua-
tion to be the valuation you started with:

sage: ww = w.augmentation(x, 2)
sage: ww._base_valuation is v
True

create_key(base_valuation, phi, mu, check=True)
Create a key which uniquely identifies the valuation over base_valuation which sends phi to mu.

Note

The uniqueness that this factory provides is not why we chose to use a factory. However, it makes pickling
and equality checks much easier. At the same time, going through a factory makes it easier to enforce
that all instances correctly inherit methods from the parent Hom space.

create_object(version, key)
Create the augmented valuation represented by key.

class sage.rings.valuation.augmented_valuation.AugmentedValuation_base(parent, v,
phi, mu)

Bases: InductiveValuation

An augmented valuation is a discrete valuation on a polynomial ring. It extends another discrete valuation 𝑣 by
setting the valuation of a polynomial 𝑓 to the minimum of 𝑣(𝑓𝑖)𝑖𝜇 when writing 𝑓 =

∑︀
𝑖 𝑓𝑖𝜑

𝑖.

INPUT:

• v – a InductiveValuation on a polynomial ring

• phi – a key polynomial over v

• mu – a rational number such that mu > v(phi) or infinity

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K.<u> = CyclotomicField(5)
sage: R.<x> = K[]
sage: v = GaussValuation(R, K.valuation(2))
sage: w = v.augmentation(x, 1/2); w # indirect doctest
[ Gauss valuation induced by 2-adic valuation, v(x) = 1/2 ]
sage: ww = w.augmentation(x^4 + 2*x^2 + 4*u, 3); ww
[ Gauss valuation induced by 2-adic valuation, v(x) = 1/2, v(x^4 + 2*x^2 + 4*u) =␣
→˓3 ]

E()

Return the ramification index of this valuation over its underlying Gauss valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1)
sage: w.E()
1
sage: w = v.augmentation(x, 1/2)
sage: w.E()
2

F()

Return the degree of the residue field extension of this valuation over the underlying Gauss valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1)
sage: w.F()
2
sage: w = v.augmentation(x, 1/2)
sage: w.F()
1

augmentation_chain()

Return a list with the chain of augmentations down to the underlying Gauss valuation.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))

(continues on next page)

sage: w = v.augmentation(x, 1)
sage: w.augmentation_chain()
[[ Gauss valuation induced by 2-adic valuation, v(x) = 1 ],
Gauss valuation induced by 2-adic valuation]

For performance reasons, (and to simplify the underlying implementation,) trivial augmentations might get
dropped. You should not rely on augmentation_chain() to contain all the steps that you specified to
create the current valuation:

sage: ww = w.augmentation(x, 2)
sage: ww.augmentation_chain()
[[ Gauss valuation induced by 2-adic valuation, v(x) = 2 ],
Gauss valuation induced by 2-adic valuation]

change_domain(ring)
Return this valuation over ring.

EXAMPLES:

We can change the domain of an augmented valuation even if there is no coercion between rings:

sage: # needs sage.rings.number_field
sage: R.<x> = GaussianIntegers()[]
sage: v = GaussValuation(R, GaussianIntegers().valuation(2))
sage: v = v.augmentation(x, 1)
sage: v.change_domain(QQ[�x�])
[ Gauss valuation induced by 2-adic valuation, v(x) = 1 ]

element_with_valuation(s)
Create an element of minimal degree and of valuation s.

INPUT:

• s – a rational number in the value group of this valuation

OUTPUT: an element in the domain of this valuation

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: w.element_with_valuation(0)
1 + O(2^5)
sage: w.element_with_valuation(1/2)
(1 + O(2^5))*x^2 + (1 + O(2^5))*x + u + O(2^5)
sage: w.element_with_valuation(1)
2 + O(2^6)
sage: c = w.element_with_valuation(-1/2); c
(2^-1 + O(2^4))*x^2 + (2^-1 + O(2^4))*x + u*2^-1 + O(2^4)
sage: w(c)
-1/2
sage: w.element_with_valuation(1/3)
Traceback (most recent call last):
...

(continues on next page)

5.8. Augmented valuations on polynomial rings 55



Discrete Valuations and Discrete Pseudo-Valuations, Release 10.5.rc0

(continued from previous page)

ValueError: s must be in the value group of the valuation
but 1/3 is not in Additive Abelian Group generated by 1/2.

equivalence_unit(s, reciprocal=False)
Return an equivalence unit of minimal degree and valuation s.

INPUT:

• s – a rational number

• reciprocal – boolean (default: False); whether or not to return the equivalence unit as theequiv-
alence_reciprocal() of the equivalence unit of valuation -s.

OUTPUT:

A polynomial in the domain of this valuation which is_equivalence_unit() for this valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1)
sage: w.equivalence_unit(0)
1 + O(2^5)
sage: w.equivalence_unit(-4)
2^-4 + O(2)

Since an equivalence unit is of effective degree zero, 𝜑 must not divide it. Therefore, its valuation is in the
value group of the base valuation:

sage: w = v.augmentation(x, 1/2) #␣
→˓needs sage.libs.ntl
sage: w.equivalence_unit(3/2) #␣
→˓needs sage.libs.ntl
Traceback (most recent call last):
...
ValueError: 3/2 is not in the value semigroup of 2-adic valuation
sage: w.equivalence_unit(1) #␣
→˓needs sage.libs.ntl
2 + O(2^6)

An equivalence unit might not be integral, even if s >= 0:

sage: w = v.augmentation(x, 3/4) #␣
→˓needs sage.libs.ntl
sage: ww = w.augmentation(x^4 + 8, 5) #␣
→˓needs sage.libs.ntl
sage: ww.equivalence_unit(1/2) #␣
→˓needs sage.libs.ntl
(2^-1 + O(2^4))*x^2

extensions(ring)
Return the extensions of this valuation to ring.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, 1)
sage: w.extensions(GaussianIntegers().fraction_field()[�x�]) #␣
→˓needs sage.rings.number_field
[[ Gauss valuation induced by 2-adic valuation, v(x^2 + x + 1) = 1 ]]

is_gauss_valuation()

Return whether this valuation is a Gauss valuation.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, 1)

sage: w.is_gauss_valuation()
False

is_negative_pseudo_valuation()

Return whether this valuation attains −∞.

EXAMPLES:

No element in the domain of an augmented valuation can have valuation −∞, so this method always returns
False:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ))
sage: w = v.augmentation(x, infinity)
sage: w.is_negative_pseudo_valuation()
False

is_trivial()

Return whether this valuation is trivial, i.e., zero outside of zero.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, 1)
sage: w.is_trivial()
False

monic_integral_model(G)
Return a monic integral irreducible polynomial which defines the same extension of the base ring of the
domain as the irreducible polynomial G together with maps between the old and the new polynomial.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, 1)
sage: w.monic_integral_model(5*x^2 + 1/2*x + 1/4)
(Ring endomorphism of Univariate Polynomial Ring in x over Rational Field

Defn: x |--> 1/2*x,
Ring endomorphism of Univariate Polynomial Ring in x over Rational Field

(continues on next page)

Defn: x |--> 2*x,
x^2 + 1/5*x + 1/5)

psi()

Return the minimal polynomial of the residue field extension of this valuation.

OUTPUT: a polynomial in the residue ring of the base valuation

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: w.psi()
x^2 + x + u0
sage: ww = w.augmentation((x^2 + x + u)^2 + 2, 5/3)
sage: ww.psi()
x + 1

restriction(ring)
Return the restriction of this valuation to ring.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = GaussianIntegers().fraction_field()
sage: R.<x> = K[]
sage: v = GaussValuation(R, K.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, 1)
sage: w.restriction(QQ[�x�]) #␣
→˓needs sage.libs.singular
[ Gauss valuation induced by 2-adic valuation, v(x^2 + x + 1) = 1 ]

scale(scalar)
Return this valuation scaled by scalar.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, 1)
sage: 3*w # indirect doctest
[ Gauss valuation induced by 3 * 2-adic valuation, v(x^2 + x + 1) = 3 ]

uniformizer()

Return a uniformizing element for this valuation.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, 1)

sage: w.uniformizer()
2

class sage.rings.valuation.augmented_valuation.FinalAugmentedValuation(parent, v,
phi, mu)

Bases: AugmentedValuation_base, FinalInductiveValuation

An augmented valuation which can not be augmented anymore, either because it augments a trivial valuation or
because it is infinite.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ))
sage: w = v.augmentation(x, 1)

lift(F)
Return a polynomial which reduces to F.

INPUT:

• F – an element of the residue_ring()

ALGORITHM:

We simply undo the steps performed in reduce().

OUTPUT: a polynomial in the domain of the valuation with reduction F

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ))

sage: w = v.augmentation(x, 1)
sage: w.lift(1/2)
1/2

sage: w = v.augmentation(x^2 + x + 1, infinity)
sage: w.lift(w.residue_ring().gen()) #␣
→˓needs sage.rings.number_field
x

A case with non-trivial base valuation:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 10)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, infinity)
sage: w.lift(w.residue_ring().gen()) #␣
→˓needs sage.rings.number_field
(1 + O(2^10))*x

reduce(f , check=True, degree_bound=None, coefficients=None, valuations=None)
Reduce f module this valuation.

INPUT:

• f – an element in the domain of this valuation

• check – whether or not to check whether f has nonnegative valuation (default: True)

• degree_bound – an a-priori known bound on the degree of the result which can speed up the com-
putation (default: not set)

• coefficients – the coefficients off as produced bycoefficients() orNone (default: None);
this can be used to speed up the computation when the expansion of f is already known from a previous
computation.

• valuations – the valuations of coefficients or None (default: None); ignored

OUTPUT:

an element of the residue_ring() of this valuation, the reduction modulo the ideal of elements of
positive valuation

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ))

sage: w = v.augmentation(x, 1)
sage: w.reduce(x^2 + x + 1)
1

sage: w = v.augmentation(x^2 + x + 1, infinity)
sage: w.reduce(x) #␣
→˓needs sage.rings.number_field
u1

residue_ring()

Return the residue ring of this valuation, i.e., the elements of nonnegative valuation modulo the elements of
positive valuation.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ))

sage: w = v.augmentation(x, 1)
sage: w.residue_ring()
Rational Field

sage: w = v.augmentation(x^2 + x + 1, infinity)
sage: w.residue_ring() #␣
→˓needs sage.rings.number_field
Number Field in u1 with defining polynomial x^2 + x + 1

An example with a non-trivial base valuation:

sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, infinity)
sage: w.residue_ring() #␣
→˓needs sage.rings.finite_rings
Finite Field in u1 of size 2^2

Since trivial extensions of finite fields are not implemented, the resulting ring might be identical to the residue
ring of the underlying valuation:

sage: w = v.augmentation(x, infinity)
sage: w.residue_ring()
Finite Field of size 2

class sage.rings.valuation.augmented_valuation.FinalFiniteAugmentedValuation(par-
ent,
v,
phi,
mu)

Bases: FiniteAugmentedValuation, FinalAugmentedValuation

An augmented valuation which is discrete, i.e., which assigns a finite valuation to its last key polynomial, but which
can not be further augmented.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, valuations.TrivialValuation(QQ))
sage: w = v.augmentation(x, 1)

class sage.rings.valuation.augmented_valuation.FiniteAugmentedValuation(parent, v,
phi, mu)

Bases: AugmentedValuation_base, FiniteInductiveValuation

A finite augmented valuation, i.e., an augmented valuation which is discrete, or equivalently an augmented valuation
which assigns to its last key polynomial a finite valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)

lower_bound(f)
Return a lower bound of this valuation at f.

Use this method to get an approximation of the valuation of f when speed is more important than accuracy.

ALGORITHM:

The main cost of evaluation is the computation of the coefficients() of the phi()-adic expansion
of f (which often leads to coefficient bloat.) So unless phi() is trivial, we fall back to valuation which this
valuation augments since it is guaranteed to be smaller everywhere.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: w.lower_bound(x^2 + x + u)
0

simplify(f , error=None, force=False, effective_degree=None, size_heuristic_bound=32, phiadic=False)
Return a simplified version of f.

Produce an element which differs from f by an element of valuation strictly greater than the valuation of f
(or strictly greater than error if set.)

INPUT:

• f – an element in the domain of this valuation

• error – a rational, infinity, or None (default: None), the error allowed to introduce through the
simplification

• force – whether or not to simplify f even if there is heuristically no change in the coefficient size of f
expected (default: False)

• effective_degree – when set, assume that coefficients beyond effective_degree in the
phi()-adic development can be safely dropped (default: None)

• size_heuristic_bound – when force is not set, the expected factor by which the coefficients
need to shrink to perform an actual simplification (default: 32)

• phiadic – whether to simplify the coefficients in the 𝜑-adic expansion recursively. This often times
leads to huge coefficients in the 𝑥-adic expansion (default: False, i.e., use an 𝑥-adic expansion.)

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: w.simplify(x^10/2 + 1, force=True)
(u + 1)*2^-1 + O(2^4)

Check that Issue #25607 has been resolved, i.e., the coefficients in the following example are small:

sage: # needs sage.libs.ntl sage.rings.number_field
sage: R.<x> = QQ[]
sage: K.<a> = NumberField(x^3 + 6)
sage: R.<x> = K[]
sage: v = GaussValuation(R, K.valuation(2))
sage: v = v.augmentation(x, 3/2)
sage: v = v.augmentation(x^2 + 8, 13/4)
sage: v = v.augmentation(x^4 + 16*x^2 + 32*x + 64, 20/3)
sage: F.<x> = FunctionField(K)
sage: S.<y> = F[]
sage: v = F.valuation(v)
sage: G = y^2 - 2*x^5 + 8*x^3 + 80*x^2 + 128*x + 192
sage: v.mac_lane_approximants(G)
[[ Gauss valuation induced by

Valuation on rational function field induced by
[ Gauss valuation induced by 2-adic valuation, v(x) = 3/2,
v(x^2 + 8) = 13/4, v(x^4 + 16*x^2 + 32*x + 64) = 20/3 ],

v(y + 4*x + 8) = 31/8 ]]

upper_bound(f)

Return an upper bound of this valuation at f.

Use this method to get an approximation of the valuation of f when speed is more important than accuracy.

ALGORITHM:

Any entry of valuations() serves as an upper bound. However, computation of the phi()-adic ex-
pansion of f is quite costly. Therefore, we produce an upper bound on the last entry of valuations(),
namely the valuation of the leading coefficient of f plus the valuation of the appropriate power of phi().

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: w.upper_bound(x^2 + x + u)
1/2

valuations(f , coefficients=None, call_error=False)
Return the valuations of the 𝑓𝑖𝜑𝑖 in the expansion 𝑓 =

∑︀
𝑖 𝑓𝑖𝜑

𝑖.

INPUT:

• f – a polynomial in the domain of this valuation

• coefficients – the coefficients off as produced bycoefficients() orNone (default: None);
this can be used to speed up the computation when the expansion of f is already known from a previous
computation.

• call_error – whether or not to speed up the computation by assuming that the result is only used to
compute the valuation of f (default: False)

OUTPUT:

An iterator over rational numbers (or infinity) [𝑣(𝑓0), 𝑣(𝑓1𝜑), . . . ]

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: list(w.valuations( x^2 + 1 ))
[0, 1/2]
sage: ww = w.augmentation((x^2 + x + u)^2 + 2, 5/3)
sage: list(ww.valuations( ((x^2 + x + u)^2 + 2)^3 ))
[+Infinity, +Infinity, +Infinity, 5]

value_group()

Return the value group of this valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: w.value_group()
Additive Abelian Group generated by 1/2
sage: ww = w.augmentation((x^2 + x + u)^2 + 2, 5/3)
sage: ww.value_group()
Additive Abelian Group generated by 1/6

value_semigroup()

Return the value semigroup of this valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Zq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: w.value_semigroup()
Additive Abelian Semigroup generated by 1/2
sage: ww = w.augmentation((x^2 + x + u)^2 + 2, 5/3)
sage: ww.value_semigroup()
Additive Abelian Semigroup generated by 1/2, 5/3

class sage.rings.valuation.augmented_valuation.InfiniteAugmentedValuation(par-
ent, v,
phi,
mu)

Bases: FinalAugmentedValuation, InfiniteInductiveValuation

An augmented valuation which is infinite, i.e., which assigns valuation infinity to its last key polynomial (and which
can therefore not be augmented further.)

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x, infinity)

lower_bound(f)
Return a lower bound of this valuation at f.

Use this method to get an approximation of the valuation of f when speed is more important than accuracy.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, infinity)
sage: w.lower_bound(x^2 + x + u)
+Infinity

simplify(f , error=None, force=False, effective_degree=None)
Return a simplified version of f.

Produce an element which differs from f by an element of valuation strictly greater than the valuation of f
(or strictly greater than error if set.)

INPUT:

• f – an element in the domain of this valuation

• error – a rational, infinity, or None (default: None), the error allowed to introduce through the
simplification

• force – whether or not to simplify f even if there is heuristically no change in the coefficient size of f
expected (default: False)

• effective_degree – ignored; for compatibility with other simplify methods

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, infinity)
sage: w.simplify(x^10/2 + 1, force=True)
(u + 1)*2^-1 + O(2^4)

upper_bound(f)
Return an upper bound of this valuation at f.

Use this method to get an approximation of the valuation of f when speed is more important than accuracy.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, infinity)
sage: w.upper_bound(x^2 + x + u)
+Infinity

valuations(f , coefficients=None, call_error=False)
Return the valuations of the 𝑓𝑖𝜑𝑖 in the expansion 𝑓 =

∑︀
𝑖 𝑓𝑖𝜑

𝑖.

INPUT:

• f – a polynomial in the domain of this valuation

• coefficients – the coefficients off as produced bycoefficients() orNone (default: None);
this can be used to speed up the computation when the expansion of f is already known from a previous
computation.

• call_error – whether or not to speed up the computation by assuming that the result is only used to
compute the valuation of f (default: False)

OUTPUT:

An iterator over rational numbers (or infinity) [𝑣(𝑓0), 𝑣(𝑓1𝜑), . . . ]

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x, infinity)
sage: list(w.valuations(x^2 + 1))
[0, +Infinity, +Infinity]

value_group()

Return the value group of this valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)

(continues on next page)

sage: w = v.augmentation(x, infinity)
sage: w.value_group()
Additive Abelian Group generated by 1

value_semigroup()

Return the value semigroup of this valuation.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Zq(4, 5)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x, infinity)
sage: w.value_semigroup()
Additive Abelian Semigroup generated by 1

class sage.rings.valuation.augmented_valuation.NonFinalAugmentedValuation(par-
ent, v,
phi,
mu)

Bases: AugmentedValuation_base, NonFinalInductiveValuation

An augmented valuation which can be augmented further.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, 1)

lift(F , report_coefficients=False)
Return a polynomial which reduces to F.

INPUT:

• F – an element of the residue_ring()

• report_coefficients – whether to return the coefficients of the phi()-adic expansion or the
actual polynomial (default: False, i.e., return the polynomial)

OUTPUT:

A polynomial in the domain of the valuation with reduction F, monic if F is monic.

ALGORITHM:

Since this is the inverse of reduce(), we only have to go backwards through the algorithm described there.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 10)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: y = w.residue_ring().gen()
sage: u1 = w.residue_ring().base().gen()
sage: w.lift(1)

(continues on next page)

1 + O(2^10)
sage: w.lift(0)
0
sage: w.lift(u1)
(1 + O(2^10))*x
sage: w.reduce(w.lift(y)) == y
True
sage: w.reduce(w.lift(y + u1 + 1)) == y + u1 + 1
True
sage: ww = w.augmentation((x^2 + x + u)^2 + 2, 5/3)
sage: y = ww.residue_ring().gen()
sage: u2 = ww.residue_ring().base().gen()
sage: ww.reduce(ww.lift(y)) == y
True
sage: ww.reduce(ww.lift(1)) == 1
True
sage: ww.reduce(ww.lift(y + 1)) == y + 1
True

A more complicated example:

sage: # needs sage.libs.ntl
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1)
sage: ww = w.augmentation((x^2 + x + u)^2 + 2*x*(x^2 + x + u) + 4*x, 3)
sage: u = ww.residue_ring().base().gen()
sage: F = ww.residue_ring()(u); F
u2
sage: f = ww.lift(F); f
(2^-1 + O(2^9))*x^2 + (2^-1 + O(2^9))*x + u*2^-1 + O(2^9)
sage: F == ww.reduce(f)
True

lift_to_key(F , check=True)
Lift the irreducible polynomial F to a key polynomial.

INPUT:

• F – an irreducible non-constant polynomial in the residue_ring() of this valuation

• check – whether or not to check correctness of F (default: True)

OUTPUT:

A polynomial 𝑓 in the domain of this valuation which is a key polynomial for this valuation and which, for a
suitable equivalence unit 𝑅, satisfies that the reduction of 𝑅𝑓 is F

ALGORITHM:

We follow the algorithm described in Theorem 13.1 [Mac1936I] which, after a lift() of F, essentially
shifts the valuations of all terms in the 𝜑-adic expansion up and then kills the leading coefficient.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 10)
sage: S.<x> = R[]
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1/2)

(continues on next page)

sage: y = w.residue_ring().gen()
sage: f = w.lift_to_key(y + 1); f
(1 + O(2^10))*x^4 + (2 + O(2^11))*x^3 + (1 + u*2 + O(2^10))*x^2 + (u*2 + O(2^
→˓11))*x + (u + 1) + u*2 + O(2^10)
sage: w.is_key(f)
True

A more complicated example:

sage: # needs sage.libs.ntl
sage: v = GaussValuation(S)
sage: w = v.augmentation(x^2 + x + u, 1)
sage: ww = w.augmentation((x^2 + x + u)^2 + 2*x*(x^2 + x + u) + 4*x, 3)
sage: u = ww.residue_ring().base().gen()
sage: y = ww.residue_ring().gen()
sage: f = ww.lift_to_key(y^3+y+u)
sage: f.degree()
12
sage: ww.is_key(f)
True

reduce(f , check=True, degree_bound=None, coefficients=None, valuations=None)
Reduce f module this valuation.

INPUT:

• f – an element in the domain of this valuation

• check – whether or not to check whether f has nonnegative valuation (default: True)

• degree_bound – an a-priori known bound on the degree of the result which can speed up the com-
putation (default: not set)

• coefficients – the coefficients off as produced bycoefficients() orNone (default: None);
this can be used to speed up the computation when the expansion of f is already known from a previous
computation.

• valuations – the valuations of coefficients or None (default: None)

OUTPUT:

an element of the residue_ring() of this valuation, the reduction modulo the ideal of elements of
positive valuation

ALGORITHM:

We follow the algorithm given in the proof of Theorem 12.1 of [Mac1936I]: If f has positive valuation,
the reduction is simply zero. Otherwise, let 𝑓 =

∑︀
𝑓𝑖𝜑

𝑖 be the expansion of 𝑓 , as computed by coeffi-
cients(). Since the valuation is zero, the exponents 𝑖must all be multiples of 𝜏 , the index the value group
of the base valuation in the value group of this valuation. Hence, there is an equivalence_unit() 𝑄
with the same valuation as 𝜑𝜏 . Let 𝑄′ be its equivalence_reciprocal(). Now, rewrite each term
𝑓𝑖𝜑

𝑖𝜏 = (𝑓𝑖𝑄
𝑖)(𝜑𝜏𝑄−1)𝑖; it turns out that the second factor in this expression is a lift of the generator of

the residue_field(). The reduction of the first factor can be computed recursively.

EXAMPLES:

sage: # needs sage.libs.ntl
sage: R.<u> = Qq(4, 10)
sage: S.<x> = R[]

(continues on next page)

sage: v = GaussValuation(S)
sage: v.reduce(x)
x
sage: v.reduce(S(u))
u0
sage: w = v.augmentation(x^2 + x + u, 1/2)
sage: w.reduce(S.one())
1
sage: w.reduce(S(2))
0
sage: w.reduce(S(u))
u0
sage: w.reduce(x) # this gives the generator of the residue field extension␣
→˓of w over v
u1
sage: f = (x^2 + x + u)^2 / 2
sage: w.reduce(f)
x
sage: w.reduce(f + x + 1)
x + u1 + 1
sage: ww = w.augmentation((x^2 + x + u)^2 + 2, 5/3)
sage: g = ((x^2 + x + u)^2 + 2)^3 / 2^5
sage: ww.reduce(g)
x
sage: ww.reduce(f)
1
sage: ww.is_equivalent(f, 1)
True
sage: ww.reduce(f * g)
x
sage: ww.reduce(f + g)
x + 1

residue_ring()

Return the residue ring of this valuation, i.e., the elements of nonnegative valuation modulo the elements of
positive valuation.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x^2 + x + 1, 1)
sage: w.residue_ring() #␣
→˓needs sage.rings.finite_rings
Univariate Polynomial Ring in x over Finite Field in u1 of size 2^2

Since trivial valuations of finite fields are not implemented, the resulting ring might be identical to the residue
ring of the underlying valuation:

sage: w = v.augmentation(x, 1)
sage: w.residue_ring()
Univariate Polynomial Ring in x over Finite Field of size 2 (using ...)

class sage.rings.valuation.augmented_valuation.NonFinalFiniteAugmentedValuation(par-
ent,
v,
phi,
mu)

Bases: FiniteAugmentedValuation, NonFinalAugmentedValuation

An augmented valuation which is discrete, i.e., which assigns a finite valuation to its last key polynomial, and which
can be augmented further.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = v.augmentation(x, 1)

5.9 Valuations which are defined as limits of valuations.

The discrete valuation of a complete field extends uniquely to a finite field extension. This is not the case anymore for
fields which are not complete with respect to their discrete valuation. In this case, the extensions essentially correspond
to the factors of the defining polynomial of the extension over the completion. However, these factors only exist over the
completion and this makes it difficult to write down such valuations with a representation of finite length.

More specifically, let 𝑣 be a discrete valuation on 𝐾 and let 𝐿 = 𝐾[𝑥]/(𝐺) a finite extension thereof. An extension of
𝑣 to 𝐿 can be represented as a discrete pseudo-valuation 𝑤′ on𝐾[𝑥] which sends 𝐺 to infinity. However, such 𝑤′ might
not be described by an augmented valuation over a Gauss valuation anymore. Instead, we may need to
write is as a limit of augmented valuations.

The classes in this module provide the means of writing down such limits and resulting valuations on quotients.

AUTHORS:

• Julian Rüth (2016-10-19): initial version

EXAMPLES:

In this function field, the unique place of K which corresponds to the zero point has two extensions to L. The valuations
corresponding to these extensions can only be approximated:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = K.valuation(1)
sage: w = v.extensions(L); w
[[ (x - 1)-adic valuation, v(y + 1) = 1 ]-adic valuation,
[ (x - 1)-adic valuation, v(y - 1) = 1 ]-adic valuation]

The same phenomenon can be observed for valuations on number fields:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = QQ.valuation(5)
sage: w = v.extensions(L); w
[[ 5-adic valuation, v(t + 2) = 1 ]-adic valuation,
[ 5-adic valuation, v(t + 3) = 1 ]-adic valuation]

Note

We often rely on approximations of valuations even if we could represent the valuation without using a limit. This is
done to improve performance as many computations already can be done correctly with an approximation:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = K.valuation(1/x)
sage: w = v.extension(L); w
Valuation at the infinite place
sage: w._base_valuation._base_valuation._improve_approximation()
sage: w._base_valuation._base_valuation._approximation
[ Gauss valuation induced by Valuation at the infinite place,

v(y) = 1/2, v(y^2 - 1/x) = +Infinity ]

REFERENCES:

Limits of inductive valuations are discussed in [Mac1936I] and [Mac1936II]. An overview can also be found in Section
4.6 of [Rüt2014].

class sage.rings.valuation.limit_valuation.LimitValuationFactory

Bases: UniqueFactory

Return a limit valuation which sends the polynomial G to infinity and is greater than or equal than base_valu-
ation.

INPUT:

• base_valuation – a discrete (pseudo-)valuation on a polynomial ring which is a discrete valuation on
the coefficient ring which can be uniquely augmented (possibly only in the limit) to a pseudo-valuation that
sends G to infinity.

• G – a squarefree polynomial in the domain of base_valuation

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = valuations.LimitValuation(v, x)
sage: w(x)
+Infinity

create_key(base_valuation, G)
Create a key from the parameters of this valuation.

EXAMPLES:

Note that this does not normalize base_valuation in any way. It is easily possible to create the same
limit in two different ways:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = valuations.LimitValuation(v, x) # indirect doctest
sage: v = v.augmentation(x, infinity)
sage: u = valuations.LimitValuation(v, x)
sage: u == w
False

The point here is that this is not meant to be invoked from user code. But mostly from other factories which
have made sure that the parameters are normalized already.

create_object(version, key)
Create an object from key.

EXAMPLES:

sage: R.<x> = QQ[]
sage: v = GaussValuation(R, QQ.valuation(2))
sage: w = valuations.LimitValuation(v, x^2 + 1) # indirect doctest

class sage.rings.valuation.limit_valuation.LimitValuation_generic(parent,
approximation)

Bases: DiscretePseudoValuation

Base class for limit valuations.

A limit valuation is realized as an approximation of a valuation and means to improve that approximation when
necessary.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = K.valuation(0)
sage: w = v.extension(L)
sage: w._base_valuation
[ Gauss valuation induced by (x)-adic valuation, v(y) = 1/2 , … ]

The currently used approximation can be found in the _approximation field:

sage: w._base_valuation._approximation #␣
→˓needs sage.rings.function_field
[ Gauss valuation induced by (x)-adic valuation, v(y) = 1/2 ]

reduce(f , check=True)
Return the reduction of f as an element of the residue_ring().

INPUT:

• f – an element in the domain of this valuation of nonnegative valuation

• check – whether or not to check that f has nonnegative valuation (default: True)

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - (x - 1))
sage: v = K.valuation(0)
sage: w = v.extension(L)
sage: w.reduce(y) # indirect doctest
u1

class sage.rings.valuation.limit_valuation.MacLaneLimitValuation(parent,
approximation, G)

Bases: LimitValuation_generic, InfiniteDiscretePseudoValuation

A limit valuation that is a pseudo-valuation on polynomial ring 𝐾[𝑥] which sends a square-free polynomial 𝐺 to
infinity.

This uses the MacLane algorithm to compute the next element in the limit.

It starts from a first valuation approximation which has a unique augmentation that sends 𝐺 to infinity and
whose uniformizer must be a uniformizer of the limit and whose residue field must contain the residue field of the
limit.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: R.<x> = QQ[]
sage: K.<i> = QQ.extension(x^2 + 1)
sage: v = K.valuation(2)
sage: u = v._base_valuation; u
[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1/2 , … ]

element_with_valuation(s)
Return an element with valuation s.

extensions(ring)
Return the extensions of this valuation to ring.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: v = GaussianIntegers().valuation(2)
sage: u = v._base_valuation
sage: u.extensions(QQ[�x�])
[[ Gauss valuation induced by 2-adic valuation, v(x + 1) = 1/2 , … ]]

is_negative_pseudo_valuation()

Return whether this valuation attains −∞.

EXAMPLES:

For a Mac Lane limit valuation, this is never the case, so this method always returns False:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = QQ.valuation(2)
sage: u = v.extension(L)
sage: u.is_negative_pseudo_valuation()
False

lift(F)
Return a lift of F from the residue_ring() to the domain of this valuation.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]

(continues on next page)

sage: L.<y> = K.extension(y^4 - x^2 - 2*x - 1)
sage: v = K.valuation(1)
sage: w = v.extensions(L)[1]; w
[ (x - 1)-adic valuation, v(y^2 - 2) = 1 ]-adic valuation
sage: s = w.reduce(y); s
u1
sage: w.lift(s) # indirect doctest
y

lower_bound(f)
Return a lower bound of this valuation at x.

Use this method to get an approximation of the valuation of x when speed is more important than accuracy.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = QQ.valuation(2)
sage: u = v.extension(L)
sage: u.lower_bound(1024*t + 1024)
10
sage: u(1024*t + 1024)
21/2

residue_ring()

Return the residue ring of this valuation, which is always a field.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = QQ.valuation(2)
sage: w = v.extension(L)
sage: w.residue_ring()
Finite Field of size 2

restriction(ring)

Return the restriction of this valuation to ring.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = QQ.valuation(2)
sage: w = v.extension(L)
sage: w._base_valuation.restriction(K)
2-adic valuation

simplify(f , error=None, force=False)
Return a simplified version of f.

Produce an element which differs from f by an element of valuation strictly greater than the valuation of f
(or strictly greater than error if set.)

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = QQ.valuation(2)
sage: u = v.extension(L)
sage: u.simplify(t + 1024, force=True)
t

uniformizer()

Return a uniformizing element for this valuation.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = K.valuation(0)
sage: w = v.extension(L)
sage: w.uniformizer() # indirect doctest
y

upper_bound(f)
Return an upper bound of this valuation at x.

Use this method to get an approximation of the valuation of x when speed is more important than accuracy.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = QQ.valuation(2)
sage: u = v.extension(L)
sage: u.upper_bound(1024*t + 1024)
21/2
sage: u(1024*t + 1024)
21/2

value_semigroup()

Return the value semigroup of this valuation.

5.10 Valuationswhich are implemented through amap to another val-
uation

EXAMPLES:

Extensions of valuations over finite field extensions 𝐿 = 𝐾[𝑥]/(𝐺) are realized through an infinite valuation on 𝐾[𝑥]
which maps 𝐺 to infinity:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)

sage: v = K.valuation(0) #␣
→˓needs sage.rings.function_field
sage: w = v.extension(L); w #␣
→˓needs sage.rings.function_field
(x)-adic valuation

sage: w._base_valuation #␣
→˓needs sage.rings.function_field
[ Gauss valuation induced by (x)-adic valuation, v(y) = 1/2 , … ]

AUTHORS:

• Julian Rüth (2016-11-10): initial version

class sage.rings.valuation.mapped_valuation.FiniteExtensionFromInfiniteValuation(par-
ent,
base_val-
u-
a-
tion)

Bases: MappedValuation_base, DiscreteValuation

A valuation on a quotient of the form 𝐿 = 𝐾[𝑥]/(𝐺) with an irreducible 𝐺 which is internally backed by a
pseudo-valuations on𝐾[𝑥] which sends 𝐺 to infinity.

INPUT:

• parent – the containing valuation space (usually the space of discrete valuations on 𝐿)

• base_valuation – an infinite valuation on𝐾[𝑥] which takes 𝐺 to infinity

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = K.valuation(0)
sage: w = v.extension(L); w
(x)-adic valuation

lower_bound(x)
Return a lower bound of this valuation at x.

Use this method to get an approximation of the valuation of x when speed is more important than accuracy.

EXAMPLES:

76 Chapter 5. More Details



Discrete Valuations and Discrete Pseudo-Valuations, Release 10.5.rc0

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = valuations.pAdicValuation(QQ, 5)
sage: u,uu = v.extensions(L)
sage: u.lower_bound(t + 2)
0
sage: u(t + 2)
1

restriction(ring)
Return the restriction of this valuation to ring.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = valuations.pAdicValuation(QQ, 2)
sage: w = v.extension(L)
sage: w.restriction(K) is v
True

simplify(x, error=None, force=False)
Return a simplified version of x.

Produce an element which differs from x by an element of valuation strictly greater than the valuation of x
(or strictly greater than error if set.)

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = valuations.pAdicValuation(QQ, 5)
sage: u,uu = v.extensions(L)
sage: f = 125*t + 1
sage: u.simplify(f, error=u(f), force=True)
1

upper_bound(x)
Return an upper bound of this valuation at x.

Use this method to get an approximation of the valuation of x when speed is more important than accuracy.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = valuations.pAdicValuation(QQ, 5)
sage: u,uu = v.extensions(L)
sage: u.upper_bound(t + 2) >= 1
True

(continues on next page)

sage: u(t + 2)
1

class sage.rings.valuation.mapped_valuation.FiniteExtensionFromLimitValuation(par-
ent,
ap-
prox-
i-
mant,
G,
ap-
prox-
i-
mants)

Bases: FiniteExtensionFromInfiniteValuation

An extension of a valuation on a finite field extensions𝐿 = 𝐾[𝑥]/(𝐺)which is induced by an infinite limit valuation
on𝐾[𝑥].

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = K.valuation(1)
sage: w = v.extensions(L); w
[[ (x - 1)-adic valuation, v(y + 1) = 1 ]-adic valuation,
[ (x - 1)-adic valuation, v(y - 1) = 1 ]-adic valuation]

class sage.rings.valuation.mapped_valuation.MappedValuation_base(parent,
base_valuation)

Bases: DiscretePseudoValuation

A valuation which is implemented through another proxy “base” valuation.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = K.valuation(0)
sage: w = v.extension(L); w
(x)-adic valuation

element_with_valuation(s)
Return an element with valuation s.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1)
sage: v = valuations.pAdicValuation(QQ, 5)

(continues on next page)

sage: u,uu = v.extensions(L)
sage: u.element_with_valuation(1)
5

lift(F)
Lift F from the residue_field() of this valuation into its domain.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = K.valuation(2)
sage: w = v.extension(L)
sage: w.lift(w.residue_field().gen())
y

reduce(f)
Return the reduction of f in the residue_field() of this valuation.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - (x - 2))
sage: v = K.valuation(0)
sage: w = v.extension(L)
sage: w.reduce(y)
u1

residue_ring()

Return the residue ring of this valuation.

EXAMPLES:

sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1) #␣
→˓needs sage.rings.number_field
sage: v = valuations.pAdicValuation(QQ, 2)
sage: v.extension(L).residue_ring() #␣
→˓needs sage.rings.number_field
Finite Field of size 2

simplify(x, error=None, force=False)
Return a simplified version of x.

Produce an element which differs from x by an element of valuation strictly greater than the valuation of x
(or strictly greater than error if set.)

If force is not set, then expensive simplifications may be avoided.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x)
sage: v = K.valuation(0)
sage: w = v.extensions(L)[0]

As _relative_size() misses the bloated term x^32, the following term does not get simplified:

sage: w.simplify(y + x^32) #␣
→˓needs sage.rings.function_field
y + x^32

In this case the simplification can be forced but this should not happen as a default as the recursive simpli-
fication can be quite costly:

sage: w.simplify(y + x^32, force=True) #␣
→˓needs sage.rings.function_field
y

uniformizer()

Return a uniformizing element of this valuation.

EXAMPLES:

sage: K = QQ
sage: R.<t> = K[]
sage: L.<t> = K.extension(t^2 + 1) #␣
→˓needs sage.rings.number_field
sage: v = valuations.pAdicValuation(QQ, 2)
sage: v.extension(L).uniformizer() #␣
→˓needs sage.rings.number_field
t + 1

5.11 Valuations which are scaled versions of another valuation

EXAMPLES:

sage: 3*ZZ.valuation(3)
3 * 3-adic valuation

AUTHORS:

• Julian Rüth (2016-11-10): initial version

class sage.rings.valuation.scaled_valuation.ScaledValuationFactory

Bases: UniqueFactory

Return a valuation which scales the valuation base by the factor s.

EXAMPLES:

sage: 3*ZZ.valuation(2) # indirect doctest
3 * 2-adic valuation

create_key(base, s)
Create a key which uniquely identifies a valuation.

create_object(version, key)
Create a valuation from key.

class sage.rings.valuation.scaled_valuation.ScaledValuation_generic(parent,
base_valuation,
s)

Bases: DiscreteValuation

A valuation which scales another base_valuation by a finite positive factor s.

EXAMPLES:

sage: v = 3*ZZ.valuation(3); v
3 * 3-adic valuation

extensions(ring)
Return the extensions of this valuation to ring.

EXAMPLES:

sage: v = 3*ZZ.valuation(5)
sage: v.extensions(GaussianIntegers().fraction_field()) #␣
→˓needs sage.rings.number_field
[3 * [ 5-adic valuation, v(x + 2) = 1 ]-adic valuation,
3 * [ 5-adic valuation, v(x + 3) = 1 ]-adic valuation]

lift(F)
Lift F from the residue_field() of this valuation into its domain.

EXAMPLES:

sage: v = 3*ZZ.valuation(2)
sage: v.lift(1)
1

reduce(f)
Return the reduction of f in the residue_field() of this valuation.

EXAMPLES:

sage: v = 3*ZZ.valuation(2)
sage: v.reduce(1)
1

residue_ring()

Return the residue field of this valuation.

EXAMPLES:

sage: v = 3*ZZ.valuation(2)
sage: v.residue_ring()
Finite Field of size 2

restriction(ring)
Return the restriction of this valuation to ring.

EXAMPLES:

sage: v = 3*QQ.valuation(5)
sage: v.restriction(ZZ)
3 * 5-adic valuation

uniformizer()

Return a uniformizing element of this valuation.

EXAMPLES:

sage: v = 3*ZZ.valuation(2)
sage: v.uniformizer()
2

value_semigroup()

Return the value semigroup of this valuation.

EXAMPLES:

sage: v2 = QQ.valuation(2)
sage: (2*v2).value_semigroup()
Additive Abelian Semigroup generated by -2, 2

5.12 Discrete valuations on function fields

AUTHORS:

• Julian Rüth (2016-10-16): initial version

EXAMPLES:

We can create classical valuations that correspond to finite and infinite places on a rational function field:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(1); v
(x - 1)-adic valuation
sage: v = K.valuation(x^2 + 1); v
(x^2 + 1)-adic valuation
sage: v = K.valuation(1/x); v
Valuation at the infinite place

Note that we can also specify valuations which do not correspond to a place of the function field:

sage: R.<x> = QQ[]
sage: w = valuations.GaussValuation(R, QQ.valuation(2))
sage: v = K.valuation(w); v
2-adic valuation

Valuations on a rational function field can then be extended to finite extensions:

sage: v = K.valuation(x - 1); v
(x - 1)-adic valuation
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x) #␣
→˓needs sage.rings.function_field
sage: w = v.extensions(L); w #␣
→˓needs sage.rings.function_field
[[ (x - 1)-adic valuation, v(y + 1) = 1 ]-adic valuation,
[ (x - 1)-adic valuation, v(y - 1) = 1 ]-adic valuation]

REFERENCES:

An overview of some computational tools relating to valuations on function fields can be found in Section 4.6 of [Rüt2014].
Most of this was originally developed for number fields in [Mac1936I] and [Mac1936II].

class sage.rings.function_field.valuation.ClassicalFunctionFieldValuation_base(par-
ent)

Bases: DiscreteFunctionFieldValuation_base

Base class for discrete valuations on rational function fields that come from points on the projective line.

class sage.rings.function_field.valuation.DiscreteFunctionFieldValuation_base(par-
ent)

Bases: DiscreteValuation

Base class for discrete valuations on function fields.

extensions(L)
Return the extensions of this valuation to L.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(x)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x) #␣
→˓needs sage.rings.function_field
sage: v.extensions(L) #␣
→˓needs sage.rings.function_field
[(x)-adic valuation]

class sage.rings.function_field.valuation.FiniteRationalFunctionFieldValuation(par-
ent,
base_val-
u-
a-
tion)

Bases: InducedRationalFunctionFieldValuation_base, ClassicalFunctionFieldVal-
uation_base, RationalFunctionFieldValuation_base

Valuation of a finite place of a function field.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(x + 1); v # indirect doctest
(x + 1)-adic valuation

A finite place with residual degree:

sage: w = K.valuation(x^2 + 1); w
(x^2 + 1)-adic valuation

A finite place with ramification:

sage: K.<t> = FunctionField(GF(3))
sage: L.<x> = FunctionField(K)
sage: u = L.valuation(x^3 - t); u
(x^3 + 2*t)-adic valuation

A finite place with residual degree and ramification:

sage: q = L.valuation(x^6 - t); q
(x^6 + 2*t)-adic valuation

class sage.rings.function_field.valuation.FunctionFieldExtensionMappedValuation(par-
ent,
base_val-
u-
a-
tion,
to_base_val-
u-
a-
tion_do-
main,
from_base_val-
u-
a-
tion_do-
main)

Bases: FunctionFieldMappedValuationRelative_base

A valuation on a finite extensions of function fields𝐿 = 𝐾[𝑦]/(𝐺)where𝐾 is another function field which redirects
to another base_valuation on an isomorphism function field𝑀 = 𝐾[𝑦]/(𝐻).

The isomorphisms must be trivial on K.

EXAMPLES:

sage: K.<x> = FunctionField(GF(2))
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 + y + x^3) #␣
→˓needs sage.rings.function_field
sage: v = K.valuation(1/x)
sage: w = v.extension(L) #␣
→˓needs sage.rings.function_field

sage: w(x) #␣
→˓needs sage.rings.function_field
-1
sage: w(y) #␣
→˓needs sage.rings.function_field
-3/2
sage: w.uniformizer() #␣
→˓needs sage.rings.function_field
1/x^2*y

restriction(ring)
Return the restriction of this valuation to ring.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(GF(2))
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 + y + x^3)
sage: v = K.valuation(1/x)
sage: w = v.extension(L)
sage: w.restriction(K) is v
True

class sage.rings.function_field.valuation.FunctionFieldFromLimitValuation(par-
ent,
ap-
proxi-
mant,
G, ap-
proxi-
mants)

Bases: FiniteExtensionFromLimitValuation, DiscreteFunctionFieldValuation_base

A valuation on a finite extensions of function fields 𝐿 = 𝐾[𝑦]/(𝐺) where𝐾 is another function field.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - (x^2 + x + 1)) #␣
→˓needs sage.rings.function_field
sage: v = K.valuation(x - 1) # indirect doctest #␣
→˓needs sage.rings.function_field
sage: w = v.extension(L); w #␣
→˓needs sage.rings.function_field
(x - 1)-adic valuation

scale(scalar)
Return this valuation scaled by scalar.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - (x^2 + x + 1))
sage: v = K.valuation(x - 1) # indirect doctest
sage: w = v.extension(L)
sage: 3*w
3 * (x - 1)-adic valuation

class sage.rings.function_field.valuation.FunctionFieldMappedValuationRelative_base(par-
ent,
base_val-
u-
a-
tion,
to_base_val-
u-
a-
tion_do-
main,
from_base_val-
u-
a-
tion_do-
main)

Bases: FunctionFieldMappedValuation_base

A valuation on a function field which relies on a base_valuation on an isomorphic function field and which
is such that the map from and to the other function field is the identity on the constant field.

EXAMPLES:

sage: K.<x> = FunctionField(GF(2))
sage: v = K.valuation(1/x); v
Valuation at the infinite place

restriction(ring)
Return the restriction of this valuation to ring.

EXAMPLES:

sage: K.<x> = FunctionField(GF(2))
sage: K.valuation(1/x).restriction(GF(2))
Trivial valuation on Finite Field of size 2

class sage.rings.function_field.valuation.FunctionFieldMappedValuation_base(par-
ent,
base_val-
ua-
tion,
to_base_val-
ua-
tion_do-
main,
from_base_val-
ua-
tion_do-
main)

Bases: FunctionFieldValuation_base, MappedValuation_base

A valuation on a function field which relies on a base_valuation on an isomorphic function field.

EXAMPLES:

sage: K.<x> = FunctionField(GF(2))
sage: v = K.valuation(1/x); v
Valuation at the infinite place

is_discrete_valuation()

Return whether this is a discrete valuation.

EXAMPLES:

sage: # needs sage.rings.function_field
sage: K.<x> = FunctionField(QQ)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 - x^4 - 1)
sage: v = K.valuation(1/x)
sage: w0,w1 = v.extensions(L)
sage: w0.is_discrete_valuation()
True

scale(scalar)
Return this valuation scaled by scalar.

EXAMPLES:

sage: K.<x> = FunctionField(GF(2))
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 + y + x^3) #␣
→˓needs sage.rings.function_field
sage: v = K.valuation(1/x)
sage: w = v.extension(L) #␣
→˓needs sage.rings.function_field
sage: 3*w #␣
→˓needs sage.rings.function_field
3 * (x)-adic valuation (in Rational function field in x over Finite Field of␣
→˓size 2 after x |--> 1/x)

class sage.rings.function_field.valuation.FunctionFieldValuationFactory

Bases: UniqueFactory

Create a valuation on domain corresponding to prime.

INPUT:

• domain – a function field

• prime – a place of the function field, a valuation on a subring, or a valuation on another function field together
with information for isomorphisms to and from that function field

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(1); v # indirect doctest
(x - 1)-adic valuation
sage: v(x)
0
sage: v(x - 1)
1

See sage.rings.function_field.function_field.FunctionField.valuation() for fur-
ther examples.

create_key_and_extra_args(domain, prime)
Create a unique key which identifies the valuation given by prime on domain.

create_key_and_extra_args_from_place(domain, generator)
Create a unique key which identifies the valuation at the place specified by generator.

create_key_and_extra_args_from_valuation(domain, valuation)
Create a unique key which identifies the valuation which extends valuation.

create_key_and_extra_args_from_valuation_on_isomorphic_field(domain, valuation,
to_valuation_do-
main,
from_valuation_do-
main)

Create a unique key which identifies the valuation which is valuation after mapping through to_val-
uation_domain.

create_object(version, key, **extra_args)
Create the valuation specified by key.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: R.<x> = QQ[]
sage: w = valuations.GaussValuation(R, QQ.valuation(2))
sage: v = K.valuation(w); v # indirect doctest
2-adic valuation

class sage.rings.function_field.valuation.FunctionFieldValuation_base(parent)
Bases: DiscretePseudoValuation

Abstract base class for any discrete (pseudo-)valuation on a function field.

class sage.rings.function_field.valuation.InducedRationalFunctionFieldValuation_base(par-
ent,
base_val-
u-
a-
tion)

Bases: FunctionFieldValuation_base

Base class for function field valuation induced by a valuation on the underlying polynomial ring.

extensions(L)
Return all extensions of this valuation to L which has a larger constant field than the domain of this valuation.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(x^2 + 1)
sage: L.<x> = FunctionField(GaussianIntegers().fraction_field())
sage: v.extensions(L) # indirect doctest
[(x - I)-adic valuation, (x + I)-adic valuation]

lift(F)
Return a lift of F to the domain of this valuation such that reduce() returns the original element.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(x)
sage: v.lift(0)
0
sage: v.lift(1)
1

reduce(f)

Return the reduction of f in residue_ring().

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(x^2 + 1)
sage: v.reduce(x) #␣
→˓needs sage.rings.number_field
u1

residue_ring()

Return the residue field of this valuation.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: K.valuation(x).residue_ring()
Rational Field

restriction(ring)
Return the restriction of this valuation to ring.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: K.valuation(x).restriction(QQ)
Trivial valuation on Rational Field

simplify(f , error=None, force=False)
Return a simplified version of f.

Produce an element which differs from f by an element of valuation strictly greater than the valuation of f
(or strictly greater than error if set.)

If force is not set, then expensive simplifications may be avoided.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(2)
sage: f = (x + 1)/(x - 1)

As the coefficients of this fraction are small, we do not simplify as this could be very costly in some cases:

sage: v.simplify(f)
(x + 1)/(x - 1)

However, simplification can be forced:

sage: v.simplify(f, force=True)
3

uniformizer()

Return a uniformizing element for this valuation.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: K.valuation(x).uniformizer()
x

value_group()

Return the value group of this valuation.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: K.valuation(x).value_group()
Additive Abelian Group generated by 1

class sage.rings.function_field.valuation.InfiniteRationalFunctionFieldValuation(par-
ent)

Bases: FunctionFieldMappedValuationRelative_base, RationalFunctionFieldValua-
tion_base, ClassicalFunctionFieldValuation_base

Valuation of the infinite place of a function field.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: v = K.valuation(1/x) # indirect doctest

class sage.rings.function_field.valuation.NonClassicalRationalFunctionFieldValuation(par-
ent,
base_val-
u-
a-
tion)

Bases: InducedRationalFunctionFieldValuation_base, RationalFunctionFieldValu-
ation_base

Valuation induced by a valuation on the underlying polynomial ring which is non-classical.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: v = GaussValuation(QQ[�x�], QQ.valuation(2))
sage: w = K.valuation(v); w # indirect doctest
2-adic valuation

residue_ring()

Return the residue field of this valuation.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: v = valuations.GaussValuation(QQ[�x�], QQ.valuation(2))
sage: w = K.valuation(v)
sage: w.residue_ring()
Rational function field in x over Finite Field of size 2

sage: R.<x> = QQ[]
sage: vv = v.augmentation(x, 1)
sage: w = K.valuation(vv)
sage: w.residue_ring()
Rational function field in x over Finite Field of size 2

sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 + 2*x) #␣
→˓needs sage.rings.function_field
sage: w.extension(L).residue_ring() #␣
→˓needs sage.rings.function_field
Function field in u2 defined by u2^2 + x

class sage.rings.function_field.valuation.RationalFunctionFieldMappedValuation(par-
ent,
base_val-
u-
a-
tion,
to_base_val-
u-
a-
tion_doain,
from_base_val-
u-
a-
tion_do-
main)

Bases: FunctionFieldMappedValuationRelative_base, RationalFunctionFieldValua-
tion_base

Valuation on a rational function field that is implemented after a map to an isomorphic rational function field.

EXAMPLES:

sage: K.<x> = FunctionField(QQ)
sage: R.<x> = QQ[]
sage: w = GaussValuation(R, QQ.valuation(2)).augmentation(x, 1)
sage: w = K.valuation(w)
sage: v = K.valuation((w, K.hom([~K.gen()]), K.hom([~K.gen()]))); v
Valuation on rational function field induced by
[ Gauss valuation induced by 2-adic valuation, v(x) = 1 ]
(in Rational function field in x over Rational Field after x |--> 1/x)

class sage.rings.function_field.valuation.RationalFunctionFieldValuation_base(par-
ent)

Bases: FunctionFieldValuation_base

Base class for valuations on rational function fields.

element_with_valuation(s)

Return an element with valuation s.

EXAMPLES:

sage: # needs sage.rings.number_field
sage: x = polygen(ZZ, �x�)
sage: K.<a> = NumberField(x^3 + 6)
sage: v = K.valuation(2)
sage: R.<x> = K[]
sage: w = GaussValuation(R, v).augmentation(x, 1/123)
sage: K.<x> = FunctionField(K)
sage: w = w.extension(K)
sage: w.element_with_valuation(122/123)
2/x
sage: w.element_with_valuation(1)
2

5.13 𝑝-adic Valuations on Number Fields and Their Subrings and
Completions

EXAMPLES:

sage: ZZ.valuation(2)
2-adic valuation
sage: QQ.valuation(3)
3-adic valuation
sage: CyclotomicField(5).valuation(5) #␣
→˓needs sage.rings.number_field
5-adic valuation
sage: GaussianIntegers().valuation(7) #␣
→˓needs sage.rings.number_field
7-adic valuation
sage: Zp(11).valuation()
11-adic valuation

These valuations can then, e.g., be used to compute approximate factorizations in the completion of a ring:

sage: v = ZZ.valuation(2)
sage: R.<x> = ZZ[]
sage: f = x^5 + x^4 + x^3 + x^2 + x - 1
sage: v.montes_factorization(f, required_precision=20) #␣
→˓needs sage.geometry.polyhedron
(x + 676027) * (x^4 + 372550*x^3 + 464863*x^2 + 385052*x + 297869)

AUTHORS:

• Julian Rüth (2013-03-16): initial version

REFERENCES:

The theory used here was originally developed in [Mac1936I] and [Mac1936II]. An overview can also be found in Chapter
4 of [Rüt2014].

class sage.rings.padics.padic_valuation.PadicValuationFactory

Bases: UniqueFactory

Create a prime-adic valuation on R.

INPUT:

• R – a subring of a number field or a subring of a local field in characteristic zero

• prime – a prime that does not split, a discrete (pseudo-)valuation, a fractional ideal, or None (default:
None)

EXAMPLES:

For integers and rational numbers, prime is just a prime of the integers:

sage: valuations.pAdicValuation(ZZ, 3)
3-adic valuation

sage: valuations.pAdicValuation(QQ, 3)
3-adic valuation

prime may be None for local rings:

sage: valuations.pAdicValuation(Qp(2))
2-adic valuation

sage: valuations.pAdicValuation(Zp(2))
2-adic valuation

But it must be specified in all other cases:

sage: valuations.pAdicValuation(ZZ)
Traceback (most recent call last):
...
ValueError: prime must be specified for this ring

It can sometimes be beneficial to define a number field extension as a quotient of a polynomial ring (since number
field extensions always compute an absolute polynomial defining the extension which can be very costly):

sage: # needs sage.rings.number_field
sage: R.<x> = QQ[]
sage: K.<a> = NumberField(x^2 + 1)
sage: R.<x> = K[]
sage: L.<b> = R.quo(x^2 + a)
sage: valuations.pAdicValuation(L, 2)
2-adic valuation

See also

NumberField_generic.valuation(), Order.valuation(), pAdicGeneric.
valuation(), RationalField.valuation(), IntegerRing_class.valuation().

create_key_and_extra_args(R, prime=None, approximants=None)
Create a unique key identifying the valuation of R with respect to prime.

EXAMPLES:

sage: QQ.valuation(2) # indirect doctest
2-adic valuation

create_key_and_extra_args_for_number_field(R, prime, approximants)
Create a unique key identifying the valuation of R with respect to prime.

EXAMPLES:

sage: GaussianIntegers().valuation(2) # indirect doctest #␣
→˓needs sage.rings.number_field
2-adic valuation

create_key_and_extra_args_for_number_field_from_ideal(R, I , prime)
Create a unique key identifying the valuation of R with respect to I.

Note

prime, the original parameter that was passed to create_key_and_extra_args(), is only used
to provide more meaningful error messages

EXAMPLES:

sage: # needs sage.rings.number_field
sage: GaussianIntegers().valuation(GaussianIntegers().number_field().
→˓fractional_ideal(2)) # indirect doctest
2-adic valuation

create_key_and_extra_args_for_number_field_from_valuation(R, v, prime,
approximants)

Create a unique key identifying the valuation of R with respect to v.

Note

prime, the original parameter that was passed to create_key_and_extra_args(), is only used
to provide more meaningful error messages

EXAMPLES:

sage: GaussianIntegers().valuation(ZZ.valuation(2)) # indirect doctest #␣
→˓needs sage.rings.number_field
2-adic valuation

create_key_for_integers(R, prime)
Create a unique key identifying the valuation of R with respect to prime.

EXAMPLES:

sage: QQ.valuation(2) # indirect doctest
2-adic valuation

create_key_for_local_ring(R, prime)
Create a unique key identifying the valuation of R with respect to prime.

EXAMPLES:

sage: Qp(2).valuation() # indirect doctest
2-adic valuation

create_object(version, key, **extra_args)
Create a 𝑝-adic valuation from key.

EXAMPLES:

sage: ZZ.valuation(5) # indirect doctest
5-adic valuation

class sage.rings.padics.padic_valuation.pAdicFromLimitValuation(parent, approximant,
G, approximants)

Bases: FiniteExtensionFromLimitValuation, pAdicValuation_base

A 𝑝-adic valuation on a number field or a subring thereof, i.e., a valuation that extends the 𝑝-adic valuation on the
integers.

EXAMPLES:

sage: v = GaussianIntegers().valuation(3); v #␣
→˓needs sage.rings.number_field
3-adic valuation

extensions(ring)
Return the extensions of this valuation to ring.

EXAMPLES:

sage: v = GaussianIntegers().valuation(3) #␣
→˓needs sage.rings.number_field
sage: v.extensions(v.domain().fraction_field()) #␣
→˓needs sage.rings.number_field
[3-adic valuation]

class sage.rings.padics.padic_valuation.pAdicValuation_base(parent, p)
Bases: DiscreteValuation

Abstract base class for 𝑝-adic valuations.

INPUT:

• ring – an integral domain

• p – a rational prime over which this valuation lies, not necessarily a uniformizer for the valuation

EXAMPLES:

sage: ZZ.valuation(3)
3-adic valuation

sage: QQ.valuation(5)
5-adic valuation

For ApA-adic rings, AApAA has to match the ApA of the ring. ::

sage: v = valuations.pAdicValuation(Zp(3), 2); v
Traceback (most recent call last):
...
ValueError: prime must be an element of positive valuation

5.13. 𝑝-adic Valuations on Number Fields and Their Subrings and Completions 95



Discrete Valuations and Discrete Pseudo-Valuations, Release 10.5.rc0

change_domain(ring)
Change the domain of this valuation to ring if possible.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.change_domain(QQ).domain()
Rational Field

extensions(ring)
Return the extensions of this valuation to ring.

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.extensions(GaussianIntegers()) #␣
→˓needs sage.rings.number_field
[2-adic valuation]

is_totally_ramified(G, include_steps=False, assume_squarefree=False)
Return whether G defines a single totally ramified extension of the completion of the domain of this valuation.

INPUT:

• G – a monic squarefree polynomial over the domain of this valuation

• include_steps – boolean (default: False); where to include the valuations produced during the
process of checking whether G is totally ramified in the return value

• assume_squarefree – boolean (default: False); whether to assume that G is square-free over
the completion of the domain of this valuation. Setting this to True can significantly improve the
performance.

ALGORITHM:

This is a simplified version of sage.rings.valuation.valuation.DiscreteValuation.
mac_lane_approximants().

EXAMPLES:

sage: # needs sage.libs.ntl
sage: k = Qp(5,4)
sage: v = k.valuation()
sage: R.<x> = k[]
sage: G = x^2 + 1
sage: v.is_totally_ramified(G) #␣
→˓needs sage.geometry.polyhedron
False
sage: G = x + 1
sage: v.is_totally_ramified(G)
True
sage: G = x^2 + 2
sage: v.is_totally_ramified(G)
False
sage: G = x^2 + 5
sage: v.is_totally_ramified(G) #␣
→˓needs sage.geometry.polyhedron
True
sage: v.is_totally_ramified(G, include_steps=True) #␣
→˓needs sage.geometry.polyhedron

(continues on next page)

(continued from previous page)

(True, [Gauss valuation induced by 5-adic valuation, [ Gauss valuation␣
→˓induced by 5-adic valuation, v((1 + O(5^4))*x) = 1/2 ]])

We consider an extension as totally ramified if its ramification index matches the degree. Hence, a trivial
extension is totally ramified:

sage: R.<x> = QQ[]
sage: v = QQ.valuation(2)
sage: v.is_totally_ramified(x)
True

is_unramified(G, include_steps=False, assume_squarefree=False)
Return whether G defines a single unramified extension of the completion of the domain of this valuation.

INPUT:

• G – a monic squarefree polynomial over the domain of this valuation

• include_steps – boolean (default: False); whether to include the approximate valuations that
were used to determine the result in the return value

• assume_squarefree – boolean (default: False); whether to assume that G is square-free over
the completion of the domain of this valuation. Setting this to True can significantly improve the
performance.

EXAMPLES:

We consider an extension as unramified if its ramification index is 1. Hence, a trivial extension is unramified:

sage: R.<x> = QQ[]
sage: v = QQ.valuation(2)
sage: v.is_unramified(x)
True

If G remains irreducible in reduction, then it defines an unramified extension:

sage: v.is_unramified(x^2 + x + 1)
True

However, even if G factors, it might define an unramified extension:

sage: v.is_unramified(x^2 + 2*x + 4) #␣
→˓needs sage.geometry.polyhedron
True

lift(x)

Lift x from the residue field to the domain of this valuation.

INPUT:

• x – an element of the residue_field()

EXAMPLES:

sage: v = ZZ.valuation(3)
sage: xbar = v.reduce(4)
sage: v.lift(xbar)
1

p()

Return the 𝑝 of this 𝑝-adic valuation.

EXAMPLES:

sage: GaussianIntegers().valuation(2).p() #␣
→˓needs sage.rings.number_field
2

reduce(x)
Reduce x modulo the ideal of elements of positive valuation.

INPUT:

• x – an element in the domain of this valuation

OUTPUT: an element of the residue_field()

EXAMPLES:

sage: v = ZZ.valuation(3)
sage: v.reduce(4)
1

restriction(ring)
Return the restriction of this valuation to ring.

EXAMPLES:

sage: v = GaussianIntegers().valuation(2) #␣
→˓needs sage.rings.number_field
sage: v.restriction(ZZ) #␣
→˓needs sage.rings.number_field
2-adic valuation

value_semigroup()

Return the value semigroup of this valuation.

EXAMPLES:

sage: v = GaussianIntegers().valuation(2) #␣
→˓needs sage.rings.number_field
sage: v.value_semigroup() #␣
→˓needs sage.rings.number_field
Additive Abelian Semigroup generated by 1/2

class sage.rings.padics.padic_valuation.pAdicValuation_int(parent, p)
Bases: pAdicValuation_base

A 𝑝-adic valuation on the integers or the rationals.

EXAMPLES:

sage: v = ZZ.valuation(3); v
3-adic valuation

inverse(x, precision)
Return an approximate inverse of x.

The element returned is such that the product differs from 1 by an element of valuation at least precision.

INPUT:

• x – an element in the domain of this valuation

• precision – a rational or infinity

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: x = 3
sage: y = v.inverse(3, 2); y
3
sage: x*y - 1
8

This might not be possible for elements of positive valuation:

sage: v.inverse(2, 2)
Traceback (most recent call last):
...
ValueError: element has no approximate inverse in this ring

Unless the precision is very small:

sage: v.inverse(2, 0)
1

residue_ring()

Return the residue field of this valuation.

EXAMPLES:

sage: v = ZZ.valuation(3)
sage: v.residue_ring()
Finite Field of size 3

simplify(x, error=None, force=False, size_heuristic_bound=32)
Return a simplified version of x.

Produce an element which differs from x by an element of valuation strictly greater than the valuation of x
(or strictly greater than error if set.)

INPUT:

• x – an element in the domain of this valuation

• error – a rational, infinity, or None (default: None), the error allowed to introduce through the
simplification

• force – ignored

• size_heuristic_bound – when force is not set, the expected factor by which the x need to
shrink to perform an actual simplification (default: 32)

EXAMPLES:

sage: v = ZZ.valuation(2)
sage: v.simplify(6, force=True)
2
sage: v.simplify(6, error=0, force=True)
0

In this example, the usual rational reconstruction misses a good answer for some moduli (because the absolute
value of the numerator is not bounded by the square root of the modulus):

sage: v = QQ.valuation(2)
sage: v.simplify(110406, error=16, force=True)
562/19
sage: Qp(2, 16)(110406).rational_reconstruction()
Traceback (most recent call last):
...
ArithmeticError: rational reconstruction of 55203 (mod 65536) does not exist

uniformizer()

Return a uniformizer of this 𝑝-adic valuation, i.e., 𝑝 as an element of the domain.

EXAMPLES:

sage: v = ZZ.valuation(3)
sage: v.uniformizer()
3

class sage.rings.padics.padic_valuation.pAdicValuation_padic(parent)
Bases: pAdicValuation_base

The 𝑝-adic valuation of a complete 𝑝-adic ring.

INPUT:

• R – a 𝑝-adic ring

EXAMPLES:

sage: v = Qp(2).valuation(); v # indirect doctest
2-adic valuation

element_with_valuation(v)
Return an element of valuation v.

INPUT:

• v – an element of the pAdicValuation_base.value_semigroup() of this valuation

EXAMPLES:

sage: R = Zp(3)
sage: v = R.valuation()
sage: v.element_with_valuation(3)
3^3 + O(3^23)

sage: # needs sage.libs.ntl
sage: K = Qp(3)
sage: R.<y> = K[]
sage: L.<y> = K.extension(y^2 + 3*y + 3)
sage: L.valuation().element_with_valuation(3/2)
y^3 + O(y^43)

lift(x)

Lift x from the residue_field() to the domain of this valuation.

INPUT:

• x – an element of the residue field of this valuation

EXAMPLES:

sage: R = Zp(3)
sage: v = R.valuation()
sage: xbar = v.reduce(R(4))
sage: v.lift(xbar)
1 + O(3^20)

reduce(x)
Reduce x modulo the ideal of elements of positive valuation.

INPUT:

• x – an element of the domain of this valuation

OUTPUT: an element of the residue_field()

EXAMPLES:

sage: R = Zp(3)
sage: Zp(3).valuation().reduce(R(4))
1

residue_ring()

Return the residue field of this valuation.

EXAMPLES:

sage: Qq(9, names=�a�).valuation().residue_ring() #␣
→˓needs sage.libs.ntl
Finite Field in a0 of size 3^2

shift(x, s)
Shift x in its expansion with respect to uniformizer() by s “digits”.

For nonnegative s, this just returns x multiplied by a power of the uniformizer 𝜋.

For negative s, it does the same but when not over a field, it drops coefficients in the 𝜋-adic expansion which
have negative valuation.

EXAMPLES:

sage: R = ZpCA(2)
sage: v = R.valuation()
sage: v.shift(R.one(), 1)
2 + O(2^20)
sage: v.shift(R.one(), -1)
O(2^19)

sage: # needs sage.libs.ntl sage.rings.padics
sage: S.<y> = R[]
sage: S.<y> = R.extension(y^3 - 2)
sage: v = S.valuation()
sage: v.shift(1, 5)
y^5 + O(y^60)

simplify(x, error=None, force=False)
Return a simplified version of x.

Produce an element which differs from x by an element of valuation strictly greater than the valuation of x
(or strictly greater than error if set.)

INPUT:

• x – an element in the domain of this valuation

• error – a rational, infinity, or None (default: None), the error allowed to introduce through the
simplification

• force – ignored

EXAMPLES:

sage: R = Zp(2)
sage: v = R.valuation()
sage: v.simplify(6)
2 + O(2^21)
sage: v.simplify(6, error=0)
0

uniformizer()

Return a uniformizer of this valuation.

EXAMPLES:

sage: v = Zp(3).valuation()
sage: v.uniformizer()
3 + O(3^21)

